Rules for integrands of the form $(f\,x)^m\,(d+e\,x^2)^q\,(a+b\,x^2+c\,x^4)^p$

1. $\int (f\,x)^m\,(e\,x^2)^q\,(a+b\,x^2+c\,x^4)^p\,dx$

x: $\int (f\,x)^m\,(e\,x^2)^q\,(a+b\,x^2+c\,x^4)^p\,dx$ when $m \in q$

Derivation: Algebraic simplification

Basis: If $m \in q$, then $(e\,x^2)^q == \frac{e^q}{f^{2q}}\,(f\,x)^{2q}$

Rule 1.2.2.4.1.1: If $m \in q$, then

$$\int (f\,x)^m\,(e\,x^2)^q\,(a+b\,x^2+c\,x^4)^p\,dx \longrightarrow \frac{e^q}{f^{2q}}\int (f\,x)^{m+2q}\,(a+b\,x^2+c\,x^4)^p\,dx$$

Program code:

```
(* Int[(f_.*x_)^m_.*(e_.*x_^2)^q_.*(a_+b_.*x_^2+c_.*x_^4)^p_.,x_Symbol] :=
  e^q/f^(2*q)*Int[(f*x)^(m+2*q)*(a+b*x^2+c*x^4)^p,x] /;
FreeQ[{a,b,c,e,f,m,p},x] && IntegerQ[q] *)
```

```
(* Int[(f_.*x_)^m_.*(e_.*x_^2)^q_.*(a_+c_.*x_^4)^p_.,x_Symbol] :=
  e^q/f^(2*q)*Int[(f*x)^(m+2*q)*(a+c*x^4)^p,x] /;
FreeQ[{a,c,e,f,m,p},x] && IntegerQ[q] *)
```

2. $\int (f\,x)^m \left(e\,x^2\right)^q \left(a+b\,x^2+c\,x^4\right)^p\,dx$ when $q \notin \mathbb{Z}$

1: $\int x^m \left(e\,x^2\right)^q \left(a+b\,x^2+c\,x^4\right)^p\,dx$ when $q \notin \mathbb{Z}\ \land\ \frac{m-1}{2} \in \mathbb{Z}$

Derivation: Integration by substitution

Basis: If $\frac{m-1}{2} \in \mathbb{Z}$, then $x^m \left(e\,x^2\right)^q == \frac{1}{e^{\frac{m-1}{2}}}\,x \left(e\,x^2\right)^{q+\frac{m-1}{2}}$

Basis: $x\,F\left[x^2\right] == \frac{1}{2}\,\text{Subst}\left[F[x],\,x,\,x^2\right]\,\partial_x x^2$

Rule 1.2.2.4.1.2.1: If $q \notin \mathbb{Z}\ \land\ \frac{m-1}{2} \in \mathbb{Z}$, then

$$\int x^m \left(e\,x^2\right)^q \left(a+b\,x^2+c\,x^4\right)^p\,dx \longrightarrow \frac{1}{2\,e^{\frac{m-1}{2}}}\,\text{Subst}\left[\int (e\,x)^{q+\frac{m-1}{2}} \left(a+b\,x+c\,x^2\right)^p\,dx,\,x,\,x^2\right]$$

Program code:

```
Int[x_^m_.*(e_.*x_^2)^q_*(a_+b_.*x_^2+c_.*x_^4)^p_.,x_Symbol] :=
  1/(2*e^((m-1)/2))*Subst[Int[(e*x)^(q+(m-1)/2)*(a+b*x+c*x^2)^p,x],x,x^2] /;
FreeQ[{a,b,c,e,p,q},x] && Not[IntegerQ[q]] && IntegerQ[(m-1)/2]
```

```
Int[x_^m_.*(e_.*x_^2)^q_*(a_+c_.*x_^4)^p_.,x_Symbol] :=
  1/(2*e^((m-1)/2))*Subst[Int[(e*x)^(q+(m-1)/2)*(a+c*x^2)^p,x],x,x^2] /;
FreeQ[{a,c,e,p,q},x] && Not[IntegerQ[q]] && IntegerQ[(m-1)/2]
```

2: $\int (f\,x)^m\,(e\,x^2)^q\,(a+b\,x^2+c\,x^4)^p\,dx$ when $q\notin\mathbb{Z}$

Derivation: Piecewise constant extraction

Basis: $\partial_x \frac{(e\,x^2)^q}{(f\,x)^{2\,q}} == 0$

Rule 1.2.2.4.1.2.2: If $q\notin\mathbb{Z}$, then

$$\int (f\,x)^m\,(e\,x^2)^q\,(a+b\,x^2+c\,x^4)^p\,dx \longrightarrow \frac{e^{\text{IntPart}[q]}\,(e\,x^2)^{\text{FracPart}[q]}}{f^{2\,\text{IntPart}[q]}\,(f\,x)^{2\,\text{FracPart}[q]}}\int (f\,x)^{m+2\,q}\,(a+b\,x^2+c\,x^4)^p\,dx$$

Program code:

```
Int[(f_.*x_)^m_.*(e_.*x_^2)^q_*(a_+b_.*x_^2+c_.*x_^4)^p_.,x_Symbol] :=
  e^IntPart[q]*(e*x^2)^FracPart[q]/(f^(2*IntPart[q])*(f*x)^(2*FracPart[q]))*Int[(f*x)^(m+2*q)*(a+b*x^2+c*x^4)^p,x] /;
FreeQ[{a,b,c,e,f,m,p,q},x] && Not[IntegerQ[q]]
```

```
Int[(f_.*x_)^m_.*(e_.*x_^2)^q_*(a_+c_.*x_^4)^p_.,x_Symbol] :=
  e^IntPart[q]*(e*x^2)^FracPart[q]/(f^(2*IntPart[q])*(f*x)^(2*FracPart[q]))*Int[(f*x)^(m+2*q)*(a+c*x^4)^p,x] /;
FreeQ[{a,c,e,f,m,p,q},x] && Not[IntegerQ[q]]
```

2: $\int x\left(d+e x^2\right)^q\left(a+b x^2+c x^4\right)^p\,dx$

Derivation: Integration by substitution

Basis: $x\,F\left[x^2\right] == \frac{1}{2}\,\text{Subst}\left[F[x],\,x,\,x^2\right]\,\partial_x x^2$

Rule 1.2.2.4.2:

$$\int x\left(d+e x^2\right)^q\left(a+b x^2+c x^4\right)^p\,dx \longrightarrow \frac{1}{2}\,\text{Subst}\left[\int(d+e x)^q\left(a+b x+c x^2\right)^p\,dx,\,x,\,x^2\right]$$

Program code:

```
Int[x_*(d_+e_.*x_^2)^q_.*(a_+b_.*x_^2+c_.*x_^4)^p_.,x_Symbol] :=
  1/2*Subst[Int[(d+e*x)^q*(a+b*x+c*x^2)^p,x],x,x^2] /;
FreeQ[{a,b,c,d,e,p,q},x]
```

```
Int[x_*(d_+e_.*x_^2)^q_.*(a_+c_.*x_^4)^p_.,x_Symbol] :=
  1/2*Subst[Int[(d+e*x)^q*(a+c*x^2)^p,x],x,x^2] /;
FreeQ[{a,c,d,e,p,q},x]
```

3. $\int (f x)^m \left(d + e x^2\right)^q \left(a + b x^2 + c x^4\right)^p \, dx$ when $b^2 - 4 a c == 0$

x: $\int (f x)^m \left(d + e x^2\right)^q \left(a + b x^2 + c x^4\right)^p \, dx$ when $b^2 - 4 a c == 0 \land p \in \mathbb{Z}$

Derivation: Algebraic simplification

Basis: If $b^2 - 4 a c == 0$, then $a + b z + c z^2 == \frac{1}{c} \left(\frac{b}{2} + c z\right)^2$

Rule 1.2.2.4.3.1: If $b^2 - 4 a c == 0 \land p \in \mathbb{Z}$, then

$$\int (f x)^m \left(d + e x^2\right)^q \left(a + b x^2 + c x^4\right)^p \, dx \longrightarrow \frac{1}{c^p} \int (f x)^m \left(d + e x^2\right)^q \left(\frac{b}{2} + c x^2\right)^{2 p} \, dx$$

Program code:

```
(* Int[(f_.*x_)^m_.*(d_+e_.*x_^2)^q_.*(a_+b_.*x_^2+c_.*x_^4)^p_.,x_Symbol] :=
  1/c^p*Int[(f*x)^m*(d+e*x^2)^q*(b/2+c*x^2)^(2*p),x] /;
FreeQ[{a,b,c,d,e,f,m,p,q},x] && EqQ[b^2-4*a*c,0] && IntegerQ[p] *)
```

2. $\int (f\,x)^m\,(d+e\,x^2)^q\,(a+b\,x^2+c\,x^4)^p\,dx$ when $b^2-4\,a\,c=0\,\wedge\,p\notin\mathbb{Z}$

1: $\int x^m\,(d+e\,x^2)^q\,(a+b\,x^2+c\,x^4)^p\,dx$ when $b^2-4\,a\,c=0\,\wedge\,p\notin\mathbb{Z}\,\wedge\,\frac{m+1}{2}\in\mathbb{Z}^+$

Derivation: Integration by substitution

Basis: If $\frac{m+1}{2}\in\mathbb{Z}$, then $x^m\,F[x^2]=\frac{1}{2}\,\text{Subst}\left[x^{\frac{m-1}{2}}\,F[x],\,x,\,x^2\right]\partial_x x^2$

Note: If this substitution rule is applied when $m\in\mathbb{Z}^-$, expressions of the form $\text{Log}\left[x^2\right]$ rather than $\text{Log}[x]$ may appear in the antiderivative.

Rule 1.2.2.4.3.2.1: If $b^2-4\,a\,c=0\,\wedge\,p\notin\mathbb{Z}\,\wedge\,\frac{m+1}{2}\in\mathbb{Z}^+$, then

$$\int x^m\,(d+e\,x^2)^q\,(a+b\,x^2+c\,x^4)^p\,dx\,\rightarrow\,\frac{1}{2}\,\text{Subst}\left[\int x^{\frac{m-1}{2}}\,(d+e\,x)^q\,(a+b\,x+c\,x^2)^p\,dx,\,x,\,x^2\right]$$

Program code:

```
Int[x_^m_.*(d_+e_.*x_^2)^q_.*(a_+b_.*x_^2+c_.*x_^4)^p_,x_Symbol] :=
  1/2*Subst[Int[x^((m-1)/2)*(d+e*x)^q*(a+b*x+c*x^2)^p,x],x,x^2] /;
FreeQ[{a,b,c,d,e,p,q},x] && EqQ[b^2-4*a*c,0] && Not[IntegerQ[p]] && IGtQ[(m+1)/2,0]
```

2: $\int (f x)^m \left(d + e x^2\right)^q \left(a + b x^2 + c x^4\right)^p \, dx$ when $b^2 - 4 a c == 0 \land p \notin \mathbb{Z}$

Derivation: Piecewise constant extraction

Basis: If $b^2 - 4 a c == 0$, then $\partial_x \frac{\left(a+b x^2+c x^4\right)^p}{\left(\frac{b}{2}+c x^2\right)^{2 p}} == 0$

Basis: If $b^2 - 4 a c == 0$, then $\frac{\left(a+b x^2+c x^4\right)^p}{\left(\frac{b}{2}+c x^2\right)^{2 p}} == \frac{\left(a+b x^2+c x^4\right)^{\text{FracPart}[p]}}{c^{\text{IntPart}[p]} \left(\frac{b}{2}+c x^2\right)^{2 \, \text{FracPart}[p]}}$

Rule 1.2.2.4.3.2.2: If $b^2 - 4 a c == 0 \land p \notin \mathbb{Z}$, then

$$\int (f x)^m \left(d + e x^2\right)^q \left(a + b x^2 + c x^4\right)^p dx \longrightarrow \frac{\left(a + b x^2 + c x^4\right)^{\text{FracPart}[p]}}{c^{\text{IntPart}[p]} \left(\frac{b}{2} + c x^2\right)^{2 \, \text{FracPart}[p]}} \int (f x)^m \left(d + e x^2\right)^q \left(\frac{b}{2} + c x^2\right)^{2 p} dx$$

–

Program code:

```
Int[(f_.*x_)^m_.*(d_+e_.*x_^2)^q_.*(a_+b_.*x_^2+c_.*x_^4)^p_,x_Symbol] :=
  (a+b*x^2+c*x^4)^FracPart[p]/(c^IntPart[p]*(b/2+c*x^2)^(2*FracPart[p]))*
    Int[(f*x)^m*(d+e*x^2)^q*(b/2+c*x^2)^(2*p),x] /;
FreeQ[{a,b,c,d,e,f,m,p,q},x] && EqQ[b^2-4*a*c,0] && Not[IntegerQ[p]]
```

4: $\int x^m \left(d+e x^2\right)^q \left(a+b x^2+c x^4\right)^p dx$ when $\frac{m-1}{2} \in \mathbb{Z}$

Derivation: Integration by substitution

Basis: If $\frac{m-1}{2} \in \mathbb{Z}$, then $x^m F\left[x^2\right] = \frac{1}{2} \text{Subst}\left[x^{\frac{m-1}{2}} F[x],\, x,\, x^2\right] \partial_x x^2$

–

Rule 1.2.2.4.4.: If $\frac{m-1}{2} \in \mathbb{Z}$, then

$$\int x^m \left(d+e x^2\right)^q \left(a+b x^2+c x^4\right)^p dx \longrightarrow \frac{1}{2} \text{Subst}\left[\int x^{\frac{m-1}{2}} (d+e x)^q \left(a+b x+c x^2\right)^p dx,\, x,\, x^2\right]$$

–

Program code:

```
Int[x_^m_.*(d_+e_.*x_^2)^q_.*(a_+b_.*x_^2+c_.*x_^4)^p_.,x_Symbol] :=
  1/2*Subst[Int[x^((m-1)/2)*(d+e*x)^q*(a+b*x+c*x^2)^p,x],x,x^2] /;
FreeQ[{a,b,c,d,e,p,q},x] && IntegerQ[(m-1)/2]
```

```
Int[x_^m_.*(d_+e_.*x_^2)^q_.*(a_+c_.*x_^4)^p_.,x_Symbol] :=
  1/2*Subst[Int[x^((m-1)/2)*(d+e*x)^q*(a+c*x^2)^p,x],x,x^2] /;
FreeQ[{a,c,d,e,p,q},x] && IntegerQ[(m+1)/2]
```

5. $\int (f x)^m (d+e x^2)^q (a+b x^2+c x^4)^p \, dx$ when $b^2-4 a c \neq 0 \land c d^2-b d e+a e^2 == 0$

1: $\int (f x)^m (d+e x^2)^q (a+b x^2+c x^4)^p \, dx$ when $b^2-4 a c \neq 0 \land c d^2-b d e+a e^2 == 0 \land p \in \mathbb{Z}$

Derivation: Algebraic simplification

Basis: If $c d^2-b d e+a e^2 == 0$, then $a+b z+c z^2 == (d+e z) \left(\frac{a}{d}+\frac{c z}{e}\right)$

Rule 1.2.2.4.5.1: If $b^2-4 a c \neq 0 \land c d^2-b d e+a e^2 == 0 \land p \in \mathbb{Z}$, then

$$\int (f x)^m (d+e x^2)^q (a+b x^2+c x^4)^p \, dx \to \int (f x)^m (d+e x^2)^{q+p} \left(\frac{a}{d}+\frac{c x^2}{e}\right)^p dx$$

Program code:

```
Int[(f_.*x_)^m_.*(d_+e_.*x_^2)^q_.*(a_+b_.*x_^2+c_.*x_^4)^p_.,x_Symbol] :=
  Int[(f*x)^m*(d+e*x^2)^(q+p)*(a/d+c/e*x^2)^p,x] /;
FreeQ[{a,b,c,d,e,f,m,q},x] && NeQ[b^2-4*a*c,0] && EqQ[c*d^2-b*d*e+a*e^2,0] && IntegerQ[p]
```

```
Int[(f_.*x_)^m_.*(d_+e_.*x_^2)^q_.*(a_+c_.*x_^4)^p_.,x_Symbol] :=
  Int[(f*x)^m*(d+e*x^2)^(q+p)*(a/d+c/e*x^2)^p,x] /;
FreeQ[{a,c,d,e,f,q,m,q},x] && EqQ[c*d^2+a*e^2,0] && IntegerQ[p]
```

2: $\int (f x)^m (d+e x^2)^q (a+b x^2+c x^4)^p \, dx$ when $b^2-4 a c \neq 0 \land c d^2-b d e+a e^2 == 0 \land p \notin \mathbb{Z}$

Derivation: Piecewise constant extraction

Basis: If $c d^2-b d e+a e^2 == 0$, then $\partial_x \frac{(a+b x^2+c x^4)^p}{(d+e x^2)^p \left(\frac{a}{d}+\frac{c x^2}{e}\right)^p} == 0$

Basis: If $c d^2-b d e+a e^2 == 0$, then $\frac{(a+b x^2+c x^4)^p}{(d+e x^2)^p \left(\frac{a}{d}+\frac{c x^2}{e}\right)^p} == \frac{(a+b x^2+c x^4)^{\text{FracPart}[p]}}{(d+e x^2)^{\text{FracPart}[p]} \left(\frac{a}{d}+\frac{c x^2}{e}\right)^{\text{FracPart}[p]}}$

Rule 1.2.2.4.5.2: If $b^2-4ac\neq 0\wedge cd^2-bde+ae^2==0\wedge p\notin\mathbb{Z}$, then

$$\int (f x)^m \left(d+e x^2\right)^q \left(a+b x^2+c x^4\right)^p dx \longrightarrow \frac{\left(a+b x^2+c x^4\right)^{\mathrm{FracPart}[p]}}{\left(d+e x^2\right)^{\mathrm{FracPart}[p]} \left(\frac{a}{d}+\frac{c x^2}{e}\right)^{\mathrm{FracPart}[p]}} \int (f x)^m \left(d+e x^2\right)^{q+p} \left(\frac{a}{d}+\frac{c x^2}{e}\right)^p dx$$

Program code:

```
Int[(f_.*x_)^m_.*(d_+e_.*x_^2)^q_*(a_+b_.*x_^2+c_.*x_^4)^p_,x_Symbol] :=
  (a+b*x^2+c*x^4)^FracPart[p]/((d+e*x^2)^FracPart[p]*(a/d+(c*x^2)/e)^FracPart[p])*
    Int[(f*x)^m*(d+e*x^2)^(q+p)*(a/d+c/e*x^2)^p,x] /;
FreeQ[{a,b,c,d,e,f,m,p,q},x] && NeQ[b^2-4*a*c,0] && EqQ[c*d^2-b*d*e+a*e^2,0] && Not[IntegerQ[p]]
```

```
Int[(f_.*x_)^m_.*(d_+e_.*x_^2)^q_*(a_+c_.*x_^4)^p_,x_Symbol] :=
  (a+c*x^4)^FracPart[p]/((d+e*x^2)^FracPart[p]*(a/d+(c*x^2)/e)^FracPart[p])*Int[(f*x)^m*(d+e*x^2)^(q+p)*(a/d+c/e*x^2)^p,x] /;
FreeQ[{a,c,d,e,f,m,p,q},x] && EqQ[c*d^2+a*e^2,0] && Not[IntegerQ[p]]
```

6. $\int (f\,x)^m \left(d+e\,x^2\right)^q \left(a+b\,x^2+c\,x^4\right)^p dx$ when $b^2-4\,a\,c\neq 0 \wedge p\in\mathbb{Z}^+$

1. $\int x^m \left(d+e\,x^2\right)^q \left(a+b\,x^2+c\,x^4\right)^p dx$ when $b^2-4\,a\,c\neq 0 \wedge p\in\mathbb{Z}^+ \wedge \left(\frac{m}{2}\mid q\right)\in\mathbb{Z} \wedge q<-1$

1: $\int x^m \left(d+e\,x^2\right)^q \left(a+b\,x^2+c\,x^4\right)^p dx$ when $b^2-4\,a\,c\neq 0 \wedge p\in\mathbb{Z}^+ \wedge \left(\frac{m}{2}\mid q\right)\in\mathbb{Z} \wedge q<-1 \wedge m>0$

Derivation: Algebraic expansion and binomial recurrence 2b

Note: If $p\in\mathbb{Z}^+ \wedge \left(\frac{m}{2}\mid q\right)\in\mathbb{Z} \wedge q<0$, then $\frac{(-d)^{m/2}}{e^{2\,p+m/2}}\sum_{k=0}^{2\,p}(-d)^k\,e^{2\,p-k}\,P_{2\,p}\left[x^2,\,k\right]$ is the coefficient of the $\left(d+e\,x^2\right)^q$ term of the partial fraction expansion of $x^m\,P_{2\,p}\left[x^2\right]\left(d+e\,x^2\right)^q$.

Note: If $p\in\mathbb{Z}^+ \wedge \left(\frac{m}{2}\mid q\right)\in\mathbb{Z} \wedge q<-1 \wedge m>0$, then $2\,e^{2\,p+m/2}\,(q+1)\,x^m\left(a+b\,x^2+c\,x^4\right)^p-(-d)^{m/2-1}\left(c\,d^2-b\,d\,e+a\,e^2\right)^p\left(d+e\,(2\,q+3)\,x^2\right)$ will be divisible by $a+b\,x^2$.

Note: In the resulting integrand the degree of the polynomial in x^2 is at most $q-1$.

– Rule 1.2.2.4.6.1.1: If $b^2-4\,a\,c\neq 0 \wedge p\in\mathbb{Z}^+ \wedge \left(\frac{m}{2}\mid q\right)\in\mathbb{Z} \wedge q<-1 \wedge m>0$, then

$$\int x^m \left(d+e\,x^2\right)^q \left(a+b\,x^2+c\,x^4\right)^p dx \longrightarrow$$

$$\frac{(-d)^{m/2}}{e^{2\,p+m/2}}\left(c\,d^2-b\,d\,e+a\,e^2\right)^p\int\left(d+e\,x^2\right)^q dx+\frac{1}{e^{2\,p+m/2}}\int\left(d+e\,x^2\right)^q\left(e^{2\,p+m/2}\,x^m\left(a+b\,x^2+c\,x^4\right)^p-(-d)^{m/2}\left(c\,d^2-b\,d\,e+a\,e^2\right)^p\right)dx \longrightarrow$$

$$\frac{(-d)^{m/2-1}\left(c\,d^2-b\,d\,e+a\,e^2\right)^p x\left(d+e\,x^2\right)^{q+1}}{2\,e^{2\,p+m/2}\,(q+1)}+$$

$$\frac{1}{2\,e^{2\,p+m/2}\,(q+1)}\int\left(d+e\,x^2\right)^{q+1}\left(\frac{1}{d+e\,x^2}\left(2\,e^{2\,p+m/2}\,(q+1)\,x^m\left(a+b\,x^2+c\,x^4\right)^p-(-d)^{m/2-1}\left(c\,d^2-b\,d\,e+a\,e^2\right)^p\left(d+e\,(2\,q+3)\,x^2\right)\right)\right)dx$$

Program code:

```
Int[x_^m_.*(d_+e_.*x_^2)^q_*(a_+b_.*x_^2+c_.*x_^4)^p_.,x_Symbol] :=
  (-d)^(m/2-1)*(c*d^2-b*d*e+a*e^2)^p*x*(d+e*x^2)^(q+1)/(2*e^(2*p+m/2)*(q+1)) +
  1/(2*e^(2*p+m/2)*(q+1))*Int[(d+e*x^2)^(q+1)*
    ExpandToSum[Together[1/(d+e*x^2)*(2*e^(2*p+m/2)*(q+1)*x^m*(a+b*x^2+c*x^4)^p-
      (-d)^(m/2-1)*(c*d^2-b*d*e+a*e^2)^p*(d+e*(2*q+3)*x^2))],x],x] /;
FreeQ[{a,b,c,d,e},x] && NeQ[b^2-4*a*c,0] && IGtQ[p,0] && ILtQ[q,-1] && IGtQ[m/2,0]
```

```
Int[x_^m_.*(d_+e_.*x_^2)^q_*(a_+c_.*x_^4)^p_.,x_Symbol] :=
  (-d)^(m/2-1)*(c*d^2+a*e^2)^p*x*(d+e*x^2)^(q+1)/(2*e^(2*p+m/2)*(q+1)) +
  1/(2*e^(2*p+m/2)*(q+1))*Int[(d+e*x^2)^(q+1)*
    ExpandToSum[Together[1/(d+e*x^2)*(2*e^(2*p+m/2)*(q+1)*x^m*(a+c*x^4)^p-
      (-d)^(m/2-1)*(c*d^2+a*e^2)^p*(d+e*(2*q+3)*x^2))],x],x] /;
FreeQ[{a,c,d,e},x] && IGtQ[p,0] && ILtQ[q,-1] && IGtQ[m/2,0]
```

2: $\int x^m \left(d+e x^2\right)^q \left(a+b x^2+c x^4\right)^p dx$ when $b^2-4 a c\neq 0 \land p\in\mathbb{Z}^+ \land \left(\frac{m}{2} \mid q\right)\in\mathbb{Z} \land q<-1 \land m<0$

Derivation: Algebraic expansion and binomial recurrence 2b

Note: If $p\in\mathbb{Z}^+ \land (m \mid q)\in\mathbb{Z} \land q<0$, then $\frac{(-d)^{m/2}}{e^{2 p+m/2}} \sum_{k=0}^{2 p} (-d)^k e^{2 p-k} P_{2 p}[x^2, k]$ is the coefficient of the $\left(d+e x^2\right)^q$ term of the partial fraction expansion of $x^m P_{2 p}[x^2] \left(d+e x^2\right)^q$.

Note: If $p\in\mathbb{Z}^+ \land (m \mid q)\in\mathbb{Z} \land q<-1 \land m<0$, then $2 (-d)^{-m/2+1} e^{2 p} (q+1) \left(a+b x^2+c x^4\right)^p - e^{-m/2} \left(c d^2-b d e+a e^2\right)^p x^{-m} \left(d+e (2 q+3) x^2\right)$ will be divisible by $a+b x^2$.

Note: In the resulting integrand the degree of the polynomial in x^2 is at most $q-1$.

Rule 1.2.2.4.6.1.2: If $b^2-4 a c\neq 0 \land p\in\mathbb{Z}^+ \land \left(\frac{m}{2} \mid q\right)\in\mathbb{Z} \land q<-1 \land m<0$, then

$$\int x^m \left(d+e x^2\right)^q \left(a+b x^2+c x^4\right)^p dx \longrightarrow$$

$$\frac{(-d)^{m/2}}{e^{2 p+m/2}} \left(c d^2-b d e+a e^2\right)^p \int \left(d+e x^2\right)^q dx +$$
$$\frac{(-d)^{m/2}}{e^{2 p}} \int x^m \left(d+e x^2\right)^q \left((-d)^{-m/2} e^{2 p} \left(a+b x^2+c x^4\right)^p - e^{-m/2} \left(c d^2-b d e+a e^2\right)^p x^{-m}\right) dx \longrightarrow$$

$$\to \frac{(-d)^{m/2-1}\left(c d^2-b d e+a e^2\right)^p x\left(d+e x^2\right)^{q+1}}{2 e^{2 p+m/2}(q+1)}+$$
$$\frac{(-d)^{m/2-1}}{2 e^{2 p}(q+1)} \int x^m\left(d+e x^2\right)^{q+1}\left(\frac{1}{d+e x^2}\left(2(-d)^{-m/2+1} e^{2 p}(q+1)\left(a+b x^2+c x^4\right)^p-e^{-m/2}\left(c d^2-b d e+a e^2\right)^p x^{-m}\left(d+e(2 q+3) x^2\right)\right)\right) dx$$

Program code:

```
Int[x_^m_*(d_+e_.*x_^2)^q_*(a_+b_.*x_^2+c_.*x_^4)^p_.,x_Symbol] :=
  (-d)^(m/2-1)*(c*d^2-b*d*e+a*e^2)^p*x*(d+e*x^2)^(q+1)/(2*e^(2*p+m/2)*(q+1)) +
  (-d)^(m/2-1)/(2*e^(2*p)*(q+1))*Int[x^m*(d+e*x^2)^(q+1)*
    ExpandToSum[Together[1/(d+e*x^2)*(2*(-d)^(-m/2+1)*e^(2*p)*(q+1)*(a+b*x^2+c*x^4)^p -
      (e^(-m/2)*(c*d^2-b*d*e+a*e^2)^p*x^(-m))*(d+e*(2*q+3)*x^2))],x],x] /;
FreeQ[{a,b,c,d,e},x] && NeQ[b^2-4*a*c,0] && IGtQ[p,0] && ILtQ[q,-1] && ILtQ[m/2,0]
```

```
Int[x_^m_*(d_+e_.*x_^2)^q_*(a_+c_.*x_^4)^p_.,x_Symbol] :=
  (-d)^(m/2-1)*(c*d^2+a*e^2)^p*x*(d+e*x^2)^(q+1)/(2*e^(2*p+m/2)*(q+1)) +
  (-d)^(m/2-1)/(2*e^(2*p)*(q+1))*Int[x^m*(d+e*x^2)^(q+1)*
    ExpandToSum[Together[1/(d+e*x^2)*(2*(-d)^(-m/2+1)*e^(2*p)*(q+1)*(a+c*x^4)^p -
      (e^(-m/2)*(c*d^2+a*e^2)^p*x^(-m))*(d+e*(2*q+3)*x^2))],x],x] /;
FreeQ[{a,c,d,e},x] && IGtQ[p,0] && ILtQ[q,-1] && ILtQ[m/2,0]
```

2: $\int (f x)^m (d+e x^2)^q (a+b x^2+c x^4)^p \, dx$ when $b^2-4 a c \neq 0 \land p \in \mathbb{Z}^+ \land q+2 \in \mathbb{Z}^+$

Derivation: Algebraic expansion

Rule 1.2.2.4.6.2: If $b^2-4 a c \neq 0 \land p \in \mathbb{Z}^+ \land q+2 \in \mathbb{Z}^+$, then

$$\int (f x)^m (d+e x^2)^q (a+b x^2+c x^4)^p \, dx \longrightarrow \int \text{ExpandIntegrand}\left[(f x)^m (d+e x^2)^q (a+b x^2+c x^4)^p, x\right] dx$$

Program code:

```
Int[(f_.*x_)^m_.*(d_+e_.*x_^2)^q_.*(a_+b_.*x_^2+c_.*x_^4)^p_.,x_Symbol] :=
  Int[ExpandIntegrand[(f*x)^m(d+e*x^2)^q*(a+b*x^2+c*x^4)^p,x],x] /;
FreeQ[{a,b,c,d,e,f,m,q},x] && NeQ[b^2-4*a*c,0] && IGtQ[p,0] && IGtQ[q,-2]
```

```
Int[(f_.*x_)^m_.*(d_+e_.*x_^2)^q_.*(a_+c_.*x_^4)^p_.,x_Symbol] :=
  Int[ExpandIntegrand[(f*x)^m(d+e*x^2)^q*(a+c*x^4)^p,x],x] /;
FreeQ[{a,c,d,e,f,m,q},x] && IGtQ[p,0] && IGtQ[q,-2]
```

3: $\int (f x)^m (d+e x^2)^q (a+b x^2+c x^4)^p \, dx$ when $b^2-4 a c \neq 0 \land p \in \mathbb{Z}^+ \land q<-1 \land m>0$

Derivation: Algebraic expansion and quadratic recurrence 2a

Rule 1.2.2.4.6.3: If $b^2-4 a c \neq 0 \land p \in \mathbb{Z}^+ \land q<-1 \land m>0$,
let $Q[x] \to \text{PolynomialQuotient}\left[(a+b x^2+c x^4)^p, d+e x^2, x\right]$ and
$R \to \text{PolynomialRemainder}\left[(a+b x^2+c x^4)^p, d+e x^2, x\right]$, then

$$\int (f x)^m (d+e x^2)^q (a+b x^2+c x^4)^p \, dx \longrightarrow$$

$$R \int (f x)^m (d+e x^2)^q \, dx + \int (f x)^{m-1} (f x)\, Q[x]\, (d+e x^2)^{q+1} \, dx \longrightarrow$$

$$-\frac{R\,(f\,x)^{m+1}\,\left(d+e\,x^2\right)^{q+1}}{2\,d\,f\,(q+1)}+\frac{f}{2\,d\,(q+1)}\int(f\,x)^{m-1}\,\left(d+e\,x^2\right)^{q+1}\,(2\,d\,(q+1)\,x\,Q[x]+R\,(m+2\,q+3)\,x)\,dx$$

–

Program code:

```
Int[(f_.*x_)^m_.*(d_+e_.*x_^2)^q_*(a_+b_.*x_^2+c_.*x_^4)^p_.,x_Symbol] :=
  With[{Qx=PolynomialQuotient[(a+b*x^2+c*x^4)^p,d+e*x^2,x],
        R=Coeff[PolynomialRemainder[(a+b*x^2+c*x^4)^p,d+e*x^2,x],x,0]},
  -R*(f*x)^(m+1)*(d+e*x^2)^(q+1)/(2*d*f*(q+1)) +
  f/(2*d*(q+1))*Int[(f*x)^(m-1)*(d+e*x^2)^(q+1)*ExpandToSum[2*d*(q+1)*x*Qx+R*(m+2*q+3)*x,x],x]] /;
FreeQ[{a,b,c,d,e,f},x] && NeQ[b^2-4*a*c,0] && IGtQ[p,0] && LtQ[q,-1] && GtQ[m,0]
```

```
Int[(f_.*x_)^m_.*(d_+e_.*x_^2)^q_*(a_+c_.*x_^4)^p_.,x_Symbol] :=
  With[{Qx=PolynomialQuotient[(a+c*x^4)^p,d+e*x^2,x],
        R=Coeff[PolynomialRemainder[(a+c*x^4)^p,d+e*x^2,x],x,0]},
  -R*(f*x)^(m+1)*(d+e*x^2)^(q+1)/(2*d*f*(q+1)) +
  f/(2*d*(q+1))*Int[(f*x)^(m-1)*(d+e*x^2)^(q+1)*ExpandToSum[2*d*(q+1)*x*Qx+R*(m+2*q+3)*x,x],x]] /;
FreeQ[{a,c,d,e,f},x] && IGtQ[p,0] && LtQ[q,-1] && GtQ[m,0]
```

4: $\int (f x)^m \left(d+e x^2\right)^q \left(a+b x^2+c x^4\right)^p dx$ when $b^2-4 a c \neq 0 \land p \in \mathbb{Z}^+ \land m<-1$

Derivation: Algebraic expansion and quadratic recurrence 3b

Rule 1.2.2.4.6.4: If $b^2-4 a c \neq 0 \land p \in \mathbb{Z}^+ \land m<-1$,

let $Q[x] \to \text{PolynomialQuotient}\left[\left(a+b x^2+c x^4\right)^p, f x, x\right]$ and $R \to \text{PolynomialRemainder}\left[\left(a+b x^2+c x^4\right)^p, f x, x\right]$,

then

$$\int (f x)^m \left(d+e x^2\right)^q \left(a+b x^2+c x^4\right)^p dx \longrightarrow$$

$$R \int (f x)^m \left(d+e x^2\right)^q dx+\int (f x)^{m+1} Q[x] \left(d+e x^2\right)^q dx \longrightarrow$$

$$\frac{R (f x)^{m+1} \left(d+e x^2\right)^{q+1}}{d f (m+1)}+\frac{1}{d f^2 (m+1)} \int (f x)^{m+2} \left(d+e x^2\right)^q \left(\frac{d f (m+1) Q[x]}{x}-e R (m+2 q+3)\right) dx$$

Program code:

```
Int[(f_.*x_)^m_*(d_+e_.*x_^2)^q_.*(a_+b_.*x_^2+c_.*x_^4)^p_.,x_Symbol] :=
  With[{Qx=PolynomialQuotient[(a+b*x^2+c*x^4)^p,f*x,x], R=PolynomialRemainder[(a+b*x^2+c*x^4)^p,f*x,x]},
  R*(f*x)^(m+1)*(d+e*x^2)^(q+1)/(d*f*(m+1)) +
  1/(d*f^2*(m+1))*Int[(f*x)^(m+2)*(d+e*x^2)^q*ExpandToSum[d*f*(m+1)*Qx/x-e*R*(m+2*q+3),x],x]] /;
FreeQ[{a,b,c,d,e,f,q},x] && NeQ[b^2-4*a*c,0] && IGtQ[p,0] && LtQ[m,-1]
```

```
Int[(f_.*x_)^m_*(d_+e_.*x_^2)^q_.*(a_+c_.*x_^4)^p_.,x_Symbol] :=
  With[{Qx=PolynomialQuotient[(a+c*x^4)^p,f*x,x], R=PolynomialRemainder[(a+c*x^4)^p,f*x,x]},
  R*(f*x)^(m+1)*(d+e*x^2)^(q+1)/(d*f*(m+1)) +
  1/(d*f^2*(m+1))*Int[(f*x)^(m+2)*(d+e*x^2)^q*ExpandToSum[d*f*(m+1)*Qx/x-e*R*(m+2*q+3),x],x]] /;
FreeQ[{a,c,d,e,f,q},x] && IGtQ[p,0] && LtQ[m,-1]
```

5: $\int (f\,x)^m \left(d+e\,x^2\right)^q \left(a+b\,x^2+c\,x^4\right)^p \,dx$ when $b^2-4\,a\,c\neq 0\,\land\,p\in\mathbb{Z}^+\land\,q\notin\mathbb{Z}\,\land\,m+4\,p+2\,q+1\neq 0$

Reference: G&R 2.104

Derivation: Algebraic expansion and binomial recurrence 3a

Rule 1.2.2.4.6.5: If $b^2-4\,a\,c\neq 0\,\land\,p\in\mathbb{Z}^+\land\,q\notin\mathbb{Z}\,\land\,m+4\,p+2\,q+1\neq 0$, then

$$\int (f\,x)^m \left(d+e\,x^2\right)^q \left(a+b\,x^2+c\,x^4\right)^p dx \longrightarrow$$

$$\frac{c^p}{f^{4\,p}}\int (f\,x)^{m+4\,p}\left(d+e\,x^2\right)^q dx+\int (f\,x)^m \left(d+e\,x^2\right)^q \left(\left(a+b\,x^2+c\,x^4\right)^p-x^{4\,p}\right) dx \longrightarrow$$

$$\frac{c^p\,(f\,x)^{m+4\,p-1}\left(d+e\,x^2\right)^{q+1}}{e\,f^{4\,p-1}\,(m+4\,p+2\,q+1)}+\frac{1}{e\,(m+4\,p+2\,q+1)}\int (f\,x)^m \left(d+e\,x^2\right)^q \left(e\,(m+4\,p+2\,q+1)\left(\left(a+b\,x^2+c\,x^4\right)^p-c^p\,x^{4\,p}\right)-d\,c^p\,(m+4\,p-1)\,x^{4\,p-2}\right) dx$$

Program code:

```
Int[(f_.*x_)^m_.*(d_+e_.*x_^2)^q_.*(a_+b_.*x_^2+c_.*x_^4)^p_.,x_Symbol] :=
  c^p*(f*x)^(m+4*p-1)*(d+e*x^2)^(q+1)/(e*f^(4*p-1)*(m+4*p+2*q+1)) +
  1/(e*(m+4*p+2*q+1))*Int[(f*x)^m*(d+e*x^2)^q*
    ExpandToSum[e*(m+4*p+2*q+1)*((a+b*x^2+c*x^4)^p-c^p*x^(4*p))-d*c^p*(m+4*p-1)*x^(4*p-2),x],x] /;
FreeQ[{a,b,c,d,e,f,m,q},x] && NeQ[b^2-4*a*c,0] && IGtQ[p,0] && Not[IntegerQ[q]] && NeQ[m+4*p+2*q+1,0]
```

```
Int[(f_.*x_)^m_.*(d_+e_.*x_^2)^q_.*(a_+c_.*x_^4)^p_.,x_Symbol] :=
  c^p*(f*x)^(m+4*p-1)*(d+e*x^2)^(q+1)/(e*f^(4*p-1)*(m+4*p+2*q+1)) +
  1/(e*(m+4*p+2*q+1))*Int[(f*x)^m*(d+e*x^2)^q*
    ExpandToSum[e*(m+4*p+2*q+1)*((a+c*x^4)^p-c^p*x^(4*p))-d*c^p*(m+4*p-1)*x^(4*p-2),x],x] /;
FreeQ[{a,c,d,e,f,m,q},x] && IGtQ[p,0] && Not[IntegerQ[q]] && NeQ[m+4*p+2*q+1,0]
```

7: $\int (f\,x)^m\,(d+e\,x^2)^q\,(a+b\,x^2+c\,x^4)^p\,dx$ when $b^2-4\,a\,c\neq 0\,\wedge\,m\in\mathbb{F}$

Derivation: Integration by substitution

Basis: If $k\in\mathbb{Z}^+$, then $(f\,x)^m\,F[x] = \frac{k}{f}\,\text{Subst}\left[x^{k\,(m+1)-1}\,F\left[\frac{x^k}{f}\right],\,x,\,(f\,x)^{1/k}\right]\,\partial_x\,(f\,x)^{1/k}$

Rule 1.2.2.4.7: If $b^2-4\,a\,c\neq 0\,\wedge\,m\in\mathbb{F}$, let $k=\text{Denominator}[m]$, then

$$\int (f\,x)^m\,(d+e\,x^2)^q\,(a+b\,x^2+c\,x^4)^p\,dx\,\longrightarrow\,\frac{k}{f}\,\text{Subst}\left[\int x^{k\,(m+1)-1}\left(d+\frac{e\,x^{2\,k}}{f^2}\right)^q\left(a+\frac{b\,x^{2\,k}}{f^2}+\frac{c\,x^{4\,k}}{f^4}\right)^p dx,\,x,\,(f\,x)^{1/k}\right]$$

Program code:

```
Int[(f_.*x_)^m_*(d_+e_.*x_^2)^q_.*(a_+b_.*x_^2+c_.*x_^4)^p_,x_Symbol] :=
  With[{k=Denominator[m]},
  k/f*Subst[Int[x^(k*(m+1)-1)*(d+e*x^(2*k)/f^2)^q*(a+b*x^(2*k)/f^k+c*x^(4*k)/f^4)^p,x],x,(f*x)^(1/k)]] /;
FreeQ[{a,b,c,d,e,f,p,q},x] && NeQ[b^2-4*a*c,0] && FractionQ[m] && IntegerQ[p]
```

```
Int[(f_.*x_)^m_*(d_+e_.*x_^2)^q_.*(a_+c_.*x_^4)^p_,x_Symbol] :=
  With[{k=Denominator[m]},
  k/f*Subst[Int[x^(k*(m+1)-1)*(d+e*x^(2*k)/f)^q*(a+c*x^(4*k)/f)^p,x],x,(f*x)^(1/k)]] /;
FreeQ[{a,c,d,e,f,p,q},x] && FractionQ[m] && IntegerQ[p]
```

8. $\int (f\,x)^m\,(d+e\,x^2)\,(a+b\,x^2+c\,x^4)^p\,dx$ when $b^2-4\,a\,c\neq 0$

   1. $\int (f\,x)^m\,(d+e\,x^2)\,(a+b\,x^2+c\,x^4)^p\,dx$ when $b^2-4\,a\,c\neq 0\,\wedge\,p>0$

      1: $\int (f\,x)^m\,(d+e\,x^2)\,(a+b\,x^2+c\,x^4)^p\,dx$ when $b^2-4\,a\,c\neq 0\,\wedge\,p>0\,\wedge\,m<-1\,\wedge\,m+2\,(2\,p+1)+1\neq 0$

Derivation: Trinomial recurrence 1a

Rule 1.2.2.4.8.1.1: If $b^2-4\,a\,c\neq 0\,\wedge\,p>0\,\wedge\,m<-1\,\wedge\,m+2\,(2\,p+1)+1\neq 0$, then

$$\int (f\,x)^m\,(d+e\,x^2)\,(a+b\,x^2+c\,x^4)^p\,dx\,\longrightarrow$$

$$\frac{(f\,x)^{m+1}\,(a+b\,x^2+c\,x^4)^p\,(d\,(4\,p+m+3)+e\,(m+1)\,x^2)}{f\,(m+1)\,(m+4\,p+3)}+$$

$$\frac{2\,p}{f^2\,(m+1)\,(m+4\,p+3)}\int (f\,x)^{m+2}\,(a+b\,x^2+c\,x^4)^{p-1}\,(2\,a\,e\,(m+1)-b\,d\,(m+4\,p+3)+(b\,e\,(m+1)-2\,c\,d\,(m+4\,p+3))\,x^2)\,dx$$

Program code:

```
Int[(f_.*x_)^m_.*(d_+e_.*x_^2)*(a_+b_.*x_^2+c_.*x_^4)^p_.,x_Symbol] :=
  (f*x)^(m+1)*(a+b*x^2+c*x^4)^p*(d*(m+4*p+3)+e*(m+1)*x^2)/(f*(m+1)*(m+4*p+3)) +
  2*p/(f^2*(m+1)*(m+4*p+3))*Int[(f*x)^(m+2)*(a+b*x^2+c*x^4)^(p-1)*
    Simp[2*a*e*(m+1)-b*d*(m+4*p+3)+(b*e*(m+1)-2*c*d*(m+4*p+3))*x^2,x],x] /;
FreeQ[{a,b,c,d,e,f},x] && NeQ[b^2-4*a*c,0] && GtQ[p,0] && LtQ[m,-1] && m+4*p+3≠0 && IntegerQ[2*p] && (IntegerQ[p] || IntegerQ[m])
```

```
Int[(f_.*x_)^m_.*(d_+e_.*x_^2)*(a_+c_.*x_^4)^p_.,x_Symbol] :=
  (f*x)^(m+1)*(a+c*x^4)^p*(d*(m+4*p+3)+e*(m+1)*x^2)/(f*(m+1)*(m+4*p+3)) +
  4*p/(f^2*(m+1)*(m+4*p+3))*Int[(f*x)^(m+2)*(a+c*x^4)^(p-1)*(a*e*(m+1)-c*d*(m+4*p+3)*x^2),x] /;
FreeQ[{a,c,d,e,f},x] && GtQ[p,0] && LtQ[m,-1] && m+4*p+3≠0 && IntegerQ[2*p] && (IntegerQ[p] || IntegerQ[m])
```

2: $\int (f x)^m (d + e x^2) (a + b x^2 + c x^4)^p \, dx$ when $b^2 - 4 a c \neq 0 \land p > 0 \land m + 4 p + 1 \neq 0 \land m + 4 p + 3 \neq 0$

Derivation: Trinomial recurrence 1b

Rule 1.2.2.4.8.1.2: If $b^2 - 4 a c \neq 0 \land p > 0 \land m + 4 p + 1 \neq 0 \land m + 4 p + 3 \neq 0$, then

$$\int (f x)^m (d + e x^2) (a + b x^2 + c x^4)^p \, dx \longrightarrow$$

$$\frac{(f x)^{m+1} (a + b x^2 + c x^4)^p (2 b e p + c d (m + 4 p + 3) + c e (4 p + m + 1) x^2)}{c f (m + 4 p + 1) (m + 4 p + 3)} +$$

$$\frac{2 p}{c (m + 4 p + 1) (m + 4 p + 3)} \int (f x)^m (a + b x^2 + c x^4)^{p-1} \cdot$$

$$(2 a c d (m + 4 p + 3) - a b e (m + 1) + (2 a c e (m + 4 p + 1) + b c d (m + 4 p + 3) - b^2 e (m + 2 p + 1)) x^2) \, dx$$

Program code:

```
Int[(f_.*x_)^m_.*(d_+e_.*x_^2)*(a_+b_.*x_^2+c_.*x_^4)^p_.,x_Symbol] :=
  (f*x)^(m+1)*(a+b*x^2+c*x^4)^p*(b*e*2*p+c*d*(m+4*p+3)+c*e*(4*p+m+1)*x^2)/
    (c*f*(4*p+m+1)*(m+4*p+3)) +
  2*p/(c*(4*p+m+1)*(m+4*p+3))*Int[(f*x)^m*(a+b*x^2+c*x^4)^(p-1)*
    Simp[2*a*c*d*(m+4*p+3)-a*b*e*(m+1)+(2*a*c*e*(4*p+m+1)+b*c*d*(m+4*p+3)-b^2*e*(m+2*p+1))*x^2,x],x] /;
FreeQ[{a,b,c,d,e,f,m},x] && NeQ[b^2-4*a*c,0] && GtQ[p,0] && NeQ[4*p+m+1,0] && NeQ[m+4*p+3,0] && IntegerQ[2*p] && (IntegerQ[p] || IntegerQ
```

```
Int[(f_.*x_)^m_.*(d_+e_.*x_^2)*(a_+c_.*x_^4)^p_.,x_Symbol] :=
  (f*x)^(m+1)*(a+c*x^4)^p*(c*d*(m+4*p+3)+c*e*(4*p+m+1)*x^2)/(c*f*(4*p+m+1)*(m+4*p+3)) +
  4*a*p/((4*p+m+1)*(m+4*p+3))*Int[(f*x)^m*(a+c*x^4)^(p-1)*Simp[d*(m+4*p+3)+e*(4*p+m+1)*x^2,x],x] /;
FreeQ[{a,c,d,e,f,m},x] && GtQ[p,0] && NeQ[4*p+m+1,0] && NeQ[m+4*p+3,0] && IntegerQ[2*p] && (IntegerQ[p] || IntegerQ[m])
```

2. $\int (f x)^m \left(d+e x^2\right) \left(a+b x^2+c x^4\right)^p dx$ when $b^2-4 a c \neq 0 \land p<-1$

1: $\int (f x)^m \left(d+e x^2\right) \left(a+b x^2+c x^4\right)^p dx$ when $b^2-4 a c \neq 0 \land p<-1 \land m>1$

Derivation: Trinomial recurrence 2a

Rule 1.2.2.4.8.2.1: If $b^2-4 a c \neq 0 \land p<-1 \land m>1$, then

$$\int (f x)^m \left(d+e x^2\right) \left(a+b x^2+c x^4\right)^p dx \longrightarrow$$

$$\frac{f (f x)^{m-1} \left(a+b x^2+c x^4\right)^{p+1} \left(b d-2 a e-(b e-2 c d) x^2\right)}{2 (p+1) \left(b^2-4 a c\right)}-$$

$$\frac{f^2}{2 (p+1) \left(b^2-4 a c\right)} \int (f x)^{m-2} \left(a+b x^2+c x^4\right)^{p+1} \left((m-1) (b d-2 a e)-(4 p+m+5) (b e-2 c d) x^2\right) dx$$

Program code:

```
Int[(f_.*x_)^m_.*(d_+e_.*x_^2)*(a_+b_.*x_^2+c_.*x_^4)^p_.,x_Symbol] :=
  f*(f*x)^(m-1)*(a+b*x^2+c*x^4)^(p+1)*(b*d-2*a*e-(b*e-2*c*d)*x^2)/(2*(p+1)*(b^2-4*a*c)) -
  f^2/(2*(p+1)*(b^2-4*a*c))*Int[(f*x)^(m-2)*(a+b*x^2+c*x^4)^(p+1)*
    Simp[(m-1)*(b*d-2*a*e)-(4*p+4+m+1)*(b*e-2*c*d)*x^2,x],x] /;
FreeQ[{a,b,c,d,e,f},x] && NeQ[b^2-4*a*c,0] && LtQ[p,-1] && GtQ[m,1] && IntegerQ[2*p] && (IntegerQ[p] || IntegerQ[m])
```

```
Int[(f_.*x_)^m_.*(d_+e_.*x_^2)*(a_+c_.*x_^4)^p_.,x_Symbol] :=
  f*(f*x)^(m-1)*(a+c*x^4)^(p+1)*(a*e-c*d*x^2)/(4*a*c*(p+1)) -
  f^2/(4*a*c*(p+1))*Int[(f*x)^(m-2)*(a+c*x^4)^(p+1)*(a*e*(m-1)-c*d*(4*p+4+m+1)*x^2),x] /;
FreeQ[{a,c,d,e,f},x] && LtQ[p,-1] && GtQ[m,1] && IntegerQ[2*p] && (IntegerQ[p] || IntegerQ[m])
```

2: $\int (f\,x)^m \left(d+e\,x^2\right) \left(a+b\,x^2+c\,x^4\right)^p \,dx$ when $b^2-4\,a\,c\neq 0\,\wedge\,p<-1$

Derivation: Trinomial recurrence 2b

Rule 1.2.2.4.8.2.2: If $b^2-4\,a\,c\neq 0\,\wedge\,p<-1$, then

$$\int (f\,x)^m \left(d+e\,x^2\right) \left(a+b\,x^2+c\,x^4\right)^p dx \,\longrightarrow$$
$$-\frac{(f\,x)^{m+1} \left(a+b\,x^2+c\,x^4\right)^{p+1} \left(d\left(b^2-2\,a\,c\right)-a\,b\,e+(b\,d-2\,a\,e)\,c\,x^2\right)}{2\,a\,f\,(p+1)\left(b^2-4\,a\,c\right)}+$$
$$\frac{1}{2\,a\,(p+1)\left(b^2-4\,a\,c\right)}\int (f\,x)^m \left(a+b\,x^2+c\,x^4\right)^{p+1}\cdot$$
$$\left(d\left(b^2\,(m+2\,p+3)-2\,a\,c\,(m+4\,(p+1)+1)\right)-a\,b\,e\,(m+1)+c\,(m+2\,(2\,p+3)+1)\,(b\,d-2\,a\,e)\,x^2\right)dx$$

Program code:

```
Int[(f_.*x_)^m_.*(d_+e_.*x_^2)*(a_+b_.*x_^2+c_.*x_^4)^p_,x_Symbol] :=
  -(f*x)^(m+1)*(a+b*x^2+c*x^4)^(p+1)*(d*(b^2-2*a*c)-a*b*e+(b*d-2*a*e)*c*x^2)/(2*a*f*(p+1)*(b^2-4*a*c)) +
  1/(2*a*(p+1)*(b^2-4*a*c))*Int[(f*x)^m*(a+b*x^2+c*x^4)^(p+1)*
     Simp[d*(b^2*(m+2*(p+1)+1)-2*a*c*(m+4*(p+1)+1))-a*b*e*(m+1)+c*(m+2*(2*p+3)+1)*(b*d-2*a*e)*x^2,x],x] /;
FreeQ[{a,b,c,d,e,f,m},x] && NeQ[b^2-4*a*c,0] && LtQ[p,-1] && IntegerQ[2*p] && (IntegerQ[p] || IntegerQ[m])
```

```
Int[(f_.*x_)^m_.*(d_+e_.*x_^2)*(a_+c_.*x_^4)^p_,x_Symbol] :=
  -(f*x)^(m+1)*(a+c*x^4)^(p+1)*(d+e*x^2)/(4*a*f*(p+1)) +
  1/(4*a*(p+1))*Int[(f*x)^m*(a+c*x^4)^(p+1)*Simp[d*(m+4*(p+1)+1)+e*(m+2*(2*p+3)+1)*x^2,x],x] /;
FreeQ[{a,c,d,e,f,m},x] && LtQ[p,-1] && IntegerQ[2*p] && (IntegerQ[p] || IntegerQ[m])
```

3: $\int (f\,x)^m\,(d+e\,x^2)\,(a+b\,x^2+c\,x^4)^p\,dx$ when $b^2-4\,a\,c\neq 0\,\wedge\,m>1\,\wedge\,m+4\,p+3\neq 0$

Derivation: Trinomial recurrence 3a

Rule 1.2.2.4.8.3: If $b^2-4\,a\,c\neq 0\,\wedge\,m>1\,\wedge\,m+4\,p+3\neq 0$, then

$$\int (f\,x)^m\,(d+e\,x^2)\,(a+b\,x^2+c\,x^4)^p\,dx\,\longrightarrow$$
$$\frac{e\,f\,(f\,x)^{m-1}\,(a+b\,x^2+c\,x^4)^{p+1}}{c\,(m+4\,p+3)}-$$
$$\frac{f^2}{c\,(m+4\,p+3)}\int (f\,x)^{m-2}\,(a+b\,x^2+c\,x^4)^p\,\left(a\,e\,(m-1)+(b\,e\,(m+2\,p+1)-c\,d\,(m+4\,p+3))\,x^2\right)\,dx$$

Program code:

```
Int[(f_.*x_)^m_.*(d_+e_.*x_^2)*(a_+b_.*x_^2+c_.*x_^4)^p_,x_Symbol] :=
  e*f*(f*x)^(m-1)*(a+b*x^2+c*x^4)^(p+1)/(c*(m+4*p+3)) -
  f^2/(c*(m+4*p+3))*Int[(f*x)^(m-2)*(a+b*x^2+c*x^4)^p*Simp[a*e*(m-1)+(b*e*(m+2*p+1)-c*d*(m+4*p+3))*x^2,x],x] /;
FreeQ[{a,b,c,d,e,f,p},x] && NeQ[b^2-4*a*c,0] && GtQ[m,1] && NeQ[m+4*p+3,0] && IntegerQ[2*p] && (IntegerQ[p] || IntegerQ[m])
```

```
Int[(f_.*x_)^m_.*(d_+e_.*x_^2)*(a_+c_.*x_^4)^p_,x_Symbol] :=
  e*f*(f*x)^(m-1)*(a+c*x^4)^(p+1)/(c*(m+4*p+3)) -
  f^2/(c*(m+4*p+3))*Int[(f*x)^(m-2)*(a+c*x^4)^p*(a*e*(m-1)-c*d*(m+4*p+3)*x^2),x] /;
FreeQ[{a,c,d,e,f,p},x] && GtQ[m,1] && NeQ[m+4*p+3,0] && IntegerQ[2*p] && (IntegerQ[p] || IntegerQ[m])
```

4: $\int (f\,x)^m\,(d+e\,x^2)\,(a+b\,x^2+c\,x^4)^p\,dx$ when $b^2-4\,a\,c\neq 0\,\wedge\,m<-1$

Derivation: Trinomial recurrence 3b

Rule 1.2.2.4.4.8.4: If $b^2-4\,a\,c\neq 0\,\wedge\,m<-1$, then

$$\int (f\,x)^m\,(d+e\,x^2)\,(a+b\,x^2+c\,x^4)^p\,dx\,\longrightarrow$$

$$\frac{d\,(f\,x)^{m+1}\,\left(a+b\,x^2+c\,x^4\right)^{p+1}}{a\,f\,(m+1)}+\frac{1}{a\,f^2\,(m+1)}\int(f\,x)^{m+2}\,\left(a+b\,x^2+c\,x^4\right)^p\,\left(a\,e\,(m+1)-b\,d\,(m+2\,p+3)-c\,d\,(m+4\,p+5)\,x^2\right)\,dx$$

– Program code:

```
Int[(f_.*x_)^m_.*(d_+e_.*x_^2)*(a_+b_.*x_^2+c_.*x_^4)^p_,x_Symbol] :=
  d*(f*x)^(m+1)*(a+b*x^2+c*x^4)^(p+1)/(a*f*(m+1)) +
  1/(a*f^2*(m+1))*Int[(f*x)^(m+2)*(a+b*x^2+c*x^4)^p*Simp[a*e*(m+1)-b*d*(m+2*p+3)-c*d*(m+4*p+5)*x^2,x],x] /;
FreeQ[{a,b,c,d,e,f,p},x] && NeQ[b^2-4*a*c,0] && LtQ[m,-1] && IntegerQ[2*p] && (IntegerQ[p] || IntegerQ[m])
```

```
Int[(f_.*x_)^m_.*(d_+e_.*x_^2)*(a_+c_.*x_^4)^p_,x_Symbol] :=
  d*(f*x)^(m+1)*(a+c*x^4)^(p+1)/(a*f*(m+1)) +
  1/(a*f^2*(m+1))*Int[(f*x)^(m+2)*(a+c*x^4)^p*(a*e*(m+1)-c*d*(m+4*p+5)*x^2),x] /;
FreeQ[{a,c,d,e,f,p},x] && LtQ[m,-1] && IntegerQ[2*p] && (IntegerQ[p] || IntegerQ[m])
```

5. $\int \frac{(f\,x)^m\,(d+e\,x^2)}{a+b\,x^2+c\,x^4}\,dx$ when $b^2-4\,a\,c\neq 0$

1: $\int \frac{(f\,x)^m\,(d+e\,x^2)}{a+b\,x^2+c\,x^4}\,dx$ when $b^2-4\,a\,c\neq 0\ \land\ c\,d^2-a\,e^2==0\ \land\ \frac{d}{e}>0\ \land\ \frac{c}{e}\,(2\,c\,d-b\,e)>0$

Derivation: Algebraic expansion

Basis: If $c\,d^2-a\,e^2=0$, let $r=\sqrt{\frac{c}{e}\,(2\,c\,d-b\,e)}$, then $\frac{d+e\,x^2}{a+b\,x^2+c\,x^4}==\frac{e}{2\,\left(\frac{c\,d}{e}+r\,x+c\,x^2\right)}+\frac{e}{2\,\left(\frac{c\,d}{e}-r\,x+c\,x^2\right)}$

Rule 1.2.2.4.8.5.1: If $b^2-4\,a\,c\neq 0\ \land\ c\,d^2-a\,e^2==0\ \land\ \frac{d}{e}>0\ \land\ \frac{c}{e}\,(2\,c\,d-b\,e)>0$, let $r=\sqrt{\frac{c}{e}\,(2\,c\,d-b\,e)}$, then

$$\int \frac{(f\,x)^m\,(d+e\,x^2)}{a+b\,x^2+c\,x^4}\,dx\ \longrightarrow\ \frac{e}{2}\int \frac{(f\,x)^m}{\frac{c\,d}{e}-r\,x+c\,x^2}\,dx+\frac{e}{2}\int \frac{(f\,x)^m}{\frac{c\,d}{e}+r\,x+c\,x^2}\,dx$$

Program code:

```
Int[(f_.*x_)^m_.*(d_+e_.*x_^2)/(a_+b_.*x_^2+c_.*x_^4), x_Symbol] :=
  With[{r=Rt[c/e*(2*c*d-b*e),2]},
  e/2*Int[(f*x)^m/(c*d/e-r*x+c*x^2),x] +
  e/2*Int[(f*x)^m/(c*d/e+r*x+c*x^2),x]] /;
FreeQ[{a,b,c,d,e,f,m},x] && NeQ[b^2-4*a*c,0] && EqQ[c*d^2-a*e^2,0] && GtQ[d/e,0] && PosQ[c/e*(2*c*d-b*e)]
```

```
Int[(f_.*x_)^m_.*(d_+e_.*x_^2)/(a_+c_.*x_^4), x_Symbol] :=
  With[{r=Rt[2*c^2*d/e,2]},
  e/2*Int[(f*x)^m/(c*d/e-r*x+c*x^2),x] +
  e/2*Int[(f*x)^m/(c*d/e+r*x+c*x^2),x]] /;
FreeQ[{a,c,d,e,f,m},x] && EqQ[c*d^2-a*e^2,0] && GtQ[d/e,0]
```

2: $\int \frac{(f\,x)^m\,(d+e\,x^2)}{a+b\,x^2+c\,x^4}\,dx$ when $b^2-4\,a\,c\neq 0$

Derivation: Algebraic expansion

- Basis: Let $q\to\sqrt{b^2-4\,a\,c}$, then $\frac{d+e\,z}{a+b\,z+c\,z^2}==\left(\frac{e}{2}+\frac{2\,c\,d-b\,e}{2\,q}\right)\frac{1}{\frac{b}{2}-\frac{q}{2}+c\,z}+\left(\frac{e}{2}-\frac{2\,c\,d-b\,e}{2\,q}\right)\frac{1}{\frac{b}{2}+\frac{q}{2}+c\,z}$

- Rule 1.2.2.4.8.5.2: If $b^2-4\,a\,c\neq 0$, let $q\to\sqrt{b^2-4\,a\,c}$, then

$$\int\frac{(f\,x)^m\,(d+e\,x^2)}{a+b\,x^2+c\,x^4}\,dx\longrightarrow\left(\frac{e}{2}+\frac{2\,c\,d-b\,e}{2\,q}\right)\int\frac{(f\,x)^m}{\frac{b}{2}-\frac{q}{2}+c\,x^2}\,dx+\left(\frac{e}{2}-\frac{2\,c\,d-b\,e}{2\,q}\right)\int\frac{(f\,x)^m}{\frac{b}{2}+\frac{q}{2}+c\,x^2}\,dx$$

Program code:

```
Int[(f_.*x_)^m_.*(d_+e_.*x_^2)/(a_+b_.*x_^2+c_.*x_^4),x_Symbol] :=
  With[{q=Rt[b^2-4*a*c,2]},
  (e/2+(2*c*d-b*e)/(2*q))*Int[(f*x)^m/(b/2-q/2+c*x^2),x] + (e/2-(2*c*d-b*e)/(2*q))*Int[(f*x)^m/(b/2+q/2+c*x^2),x]] /;
FreeQ[{a,b,c,d,e,f,m},x] && NeQ[b^2-4*a*c,0]
```

```
Int[(f_.*x_)^m_.*(d_+e_.*x_^2)/(a_+c_.*x_^4),x_Symbol] :=
  With[{q=Rt[-a*c,2]},
  -(e/2+c*d/(2*q))*Int[(f*x)^m/(q-c*x^2),x] + (e/2-c*d/(2*q))*Int[(f*x)^m/(q+c*x^2),x]] /;
FreeQ[{a,c,d,e,f,m},x]
```

9. $\int \frac{(f\,x)^m\,(d+e\,x^2)^q}{a+b\,x^2+c\,x^4}\,dx$ when $b^2-4\,a\,c\neq 0$

1. $\int \frac{(f\,x)^m\,(d+e\,x^2)^q}{a+b\,x^2+c\,x^4}\,dx$ when $b^2-4\,a\,c\neq 0\,\land\,q\in\mathbb{Z}$

1: $\int \frac{(f\,x)^m\,(d+e\,x^2)^q}{a+b\,x^2+c\,x^4}\,dx$ when $b^2-4\,a\,c\neq 0\,\land\,q\in\mathbb{Z}\,\land\,m\in\mathbb{Z}$

Derivation: Algebraic expansion

Rule 1.2.2.4.9.1.1: If $b^2-4\,a\,c\neq 0\,\land\,q\in\mathbb{Z}\,\land\,m\in\mathbb{Z}$, then

$$\int \frac{(f\,x)^m\,(d+e\,x^2)^q}{a+b\,x^2+c\,x^4}\,dx\,\longrightarrow\,\int \text{ExpandIntegrand}\left[\frac{(f\,x)^m\,(d+e\,x^2)^q}{a+b\,x^2+c\,x^4},\,x\right]dx$$

Program code:

```
Int[(f_.*x_)^m_.*(d_+e_.*x_^2)^q_./(a_+b_.*x_^2+c_.*x_^4),x_Symbol] :=
  Int[ExpandIntegrand[(f*x)^m*(d+e*x^2)^q/(a+b*x^2+c*x^4),x],x] /;
FreeQ[{a,b,c,d,e,f,m},x] && NeQ[b^2-4*a*c,0] && IntegerQ[q] && IntegerQ[m]
```

```
Int[(f_.*x_)^m_.*(d_+e_.*x_^2)^q_./(a_+c_.*x_^4),x_Symbol] :=
  Int[ExpandIntegrand[(f*x)^m*(d+e*x^2)^q/(a+c*x^4),x],x] /;
FreeQ[{a,c,d,e,f,m},x] && IntegerQ[q] && IntegerQ[m]
```

2: $\int \frac{(f\,x)^m\,(d+e\,x^2)^q}{a+b\,x^2+c\,x^4}\,dx$ when $b^2-4\,a\,c\neq 0\,\land\,q\in\mathbb{Z}\,\land\,m\notin\mathbb{Z}$

Derivation: Algebraic expansion

Rule 1.2.2.4.9.1.2: If $b^2-4\,a\,c\neq 0\,\land\,q\in\mathbb{Z}\,\land\,m\notin\mathbb{Z}$, then

$$\int \frac{(f\,x)^m\,\left(d+e\,x^2\right)^q}{a+b\,x^2+c\,x^4}\,dx \longrightarrow \int (f\,x)^m\,\text{ExpandIntegrand}\left[\frac{\left(d+e\,x^2\right)^q}{a+b\,x^2+c\,x^4},\,x\right]dx$$

Program code:

```
Int[(f_.*x_)^m_.*(d_+e_.*x_^2)^q_./(a_+b_.*x_^2+c_.*x_^4),x_Symbol] :=
  Int[ExpandIntegrand[(f*x)^m,(d+e*x^2)^q/(a+b*x^2+c*x^4),x],x] /;
FreeQ[{a,b,c,d,e,f,m},x] && NeQ[b^2-4*a*c,0] && IntegerQ[q] && Not[IntegerQ[m]]
```

```
Int[(f_.*x_)^m_.*(d_+e_.*x_^2)^q_./(a_+c_.*x_^4),x_Symbol] :=
  Int[ExpandIntegrand[(f*x)^m,(d+e*x^2)^q/(a+c*x^4),x],x] /;
FreeQ[{a,c,d,e,f,m},x] && IntegerQ[q] && Not[IntegerQ[m]]
```

2. $\int \frac{(f\,x)^m\,(d+e\,x^2)^q}{a+b\,x^2+c\,x^4}\,dx$ when $b^2-4\,a\,c\neq 0\wedge q\notin\mathbb{Z}$

   1. $\int \frac{(f\,x)^m\,(d+e\,x^2)^q}{a+b\,x^2+c\,x^4}\,dx$ when $b^2-4\,a\,c\neq 0\wedge q\notin\mathbb{Z}\wedge q>0$

      1. $\int \frac{(f\,x)^m\,(d+e\,x^2)^q}{a+b\,x^2+c\,x^4}\,dx$ when $b^2-4\,a\,c\neq 0\wedge q\notin\mathbb{Z}\wedge q>0\wedge m>1$

         1: $\int \frac{(f\,x)^m\,(d+e\,x^2)^q}{a+b\,x^2+c\,x^4}\,dx$ when $b^2-4\,a\,c\neq 0\wedge q\notin\mathbb{Z}\wedge q>0\wedge m>3$

Reference: Algebraic expansion

Basis: $\frac{d+e\,z}{a+b\,z+c\,z^2}==\frac{c\,d-b\,e+c\,e\,z}{c^2\,z^2}-\frac{a\,(c\,d-b\,e)+\left(b\,c\,d-b^2\,e+a\,c\,e\right)\,z}{c^2\,z^2\,\left(a+b\,z+c\,z^2\right)}$

Rule 1.2.2.4.9.2.1.1.1: If $b^2-4\,a\,c\neq 0\wedge q\notin\mathbb{Z}\wedge q>0\wedge m>3$, then

$$\int \frac{(f\,x)^m\,(d+e\,x^2)^q}{a+b\,x^2+c\,x^4}\,dx\longrightarrow$$
$$\frac{f^4}{c^2}\int (f\,x)^{m-4}\,\left(c\,d-b\,e+c\,e\,x^2\right)\,\left(d+e\,x^2\right)^{q-1}\,dx-\frac{f^4}{c^2}\int \frac{(f\,x)^{m-4}\,\left(d+e\,x^2\right)^{q-1}\,\left(a\,(c\,d-b\,e)+\left(b\,c\,d-b^2\,e+a\,c\,e\right)\,x^2\right)}{a+b\,x^2+c\,x^4}\,dx$$

Program code:

```
Int[(f_.*x_)^m_.*(d_.+e_.*x_^2)^q_/(a_+b_.*x_^2+c_.*x_^4),x_Symbol] :=
  f^4/c^2*Int[(f*x)^(m-4)*(c*d-b*e+c*e*x^2)*(d+e*x^2)^(q-1),x] -
  f^4/c^2*Int[(f*x)^(m-4)*(d+e*x^2)^(q-1)*Simp[a*(c*d-b*e)+(b*c*d-b^2*e+a*c*e)*x^2,x]/(a+b*x^2+c*x^4),x] /;
FreeQ[{a,b,c,d,e,f},x] && NeQ[b^2-4*a*c,0] && Not[IntegerQ[q]] && GtQ[q,0] && GtQ[m,3]
```

```
Int[(f_.*x_)^m_.*(d_.+e_.*x_^2)^q_/(a_+c_.*x_^4),x_Symbol] :=
  f^4/c*Int[(f*x)^(m-4)*(d+e*x^2)^q,x] -
  a*f^4/c*Int[(f*x)^(m-4)*(d+e*x^2)^q/(a+c*x^4),x] /;
FreeQ[{a,c,d,e,f,q},x] && Not[IntegerQ[q]] && GtQ[m,3]
```

2: $\int \frac{(f x)^m \left(d+e x^2\right)^q}{a+b x^2+c x^4} \, dx$ when $b^2-4 a c \neq 0 \land q \notin \mathbb{Z} \land q>0 \land 1<m \leq 3$

Reference: Algebraic expansion

Basis: $\frac{d+e z}{a+b z+c z^2} = \frac{e}{c z} - \frac{a e-(c d-b e) z}{c z \left(a+b z+c z^2\right)}$

Rule 1.2.2.4.9.2.1.1.2: If $b^2-4 a c \neq 0 \land q \notin \mathbb{Z} \land q>0 \land 1<m \leq 3$, then

$$\int \frac{(f x)^m \left(d+e x^2\right)^q}{a+b x^2+c x^4} \, dx \rightarrow \frac{e f^2}{c} \int (f x)^{m-2} \left(d+e x^2\right)^{q-1} \, dx - \frac{f^2}{c} \int \frac{(f x)^{m-2} \left(d+e x^2\right)^{q-1} \left(a e-(c d-b e) x^2\right)}{a+b x^2+c x^4} \, dx$$

Program code:

```
Int[(f_.*x_)^m_.*(d_.+e_.*x_^2)^q_/(a_+b_.*x_^2+c_.*x_^4),x_Symbol] :=
  e*f^2/c*Int[(f*x)^(m-2)*(d+e*x^2)^(q-1),x] -
  f^2/c*Int[(f*x)^(m-2)*(d+e*x^2)^(q-1)*Simp[a*e-(c*d-b*e)*x^2,x]/(a+b*x^2+c*x^4),x] /;
FreeQ[{a,b,c,d,e,f},x] && NeQ[b^2-4*a*c,0] && Not[IntegerQ[q]] && GtQ[q,0] && GtQ[m,1] && LeQ[m,3]
```

```
Int[(f_.*x_)^m_.*(d_.+e_.*x_^2)^q_/(a_+c_.*x_^4),x_Symbol] :=
  e*f^2/c*Int[(f*x)^(m-2)*(d+e*x^2)^(q-1),x] -
  f^2/c*Int[(f*x)^(m-2)*(d+e*x^2)^(q-1)*Simp[a*e-c*d*x^2,x]/(a+c*x^4),x] /;
FreeQ[{a,c,d,e,f},x] && Not[IntegerQ[q]] && GtQ[q,0] && GtQ[m,1] && LeQ[m,3]
```

2: $\int \frac{(f\,x)^m\,(d+e\,x^2)^q}{a+b\,x^2+c\,x^4}\,dx$ when $b^2-4\,a\,c\neq 0\,\wedge\,q\notin\mathbb{Z}\,\wedge\,q>0\,\wedge\,m<0$

Reference: Algebraic expansion

Basis: $\frac{d+e\,z}{a+b\,z+c\,z^2}==\frac{d}{a}-\frac{z\,(b\,d-a\,e+c\,d\,z)}{a\,(a+b\,z+c\,z^2)}$

Rule 1.2.2.4.9.2.1.2: If $b^2-4\,a\,c\neq 0\,\wedge\,q\notin\mathbb{Z}\,\wedge\,q>0\,\wedge\,m<0$, then

$$\int \frac{(f\,x)^m\,(d+e\,x^2)^q}{a+b\,x^2+c\,x^4}\,dx\,\longrightarrow\,\frac{d}{a}\int (f\,x)^m\,(d+e\,x^2)^{q-1}\,dx-\frac{1}{a\,f^2}\int \frac{(f\,x)^{m+2}\,(d+e\,x^2)^{q-1}\,(b\,d-a\,e+c\,d\,x^2)}{a+b\,x^2+c\,x^4}\,dx$$

Program code:

```
Int[(f_.*x_)^m_*(d_.+e_.*x_^2)^q_/(a_+b_.*x_^2+c_.*x_^4),x_Symbol] :=
  d/a*Int[(f*x)^m*(d+e*x^2)^(q-1),x] -
  1/(a*f^2)*Int[(f*x)^(m+2)*(d+e*x^2)^(q-1)*Simp[b*d-a*e+c*d*x^2,x]/(a+b*x^2+c*x^4),x] /;
FreeQ[{a,b,c,d,e,f},x] && NeQ[b^2-4*a*c,0] && Not[IntegerQ[q]] && GtQ[q,0] && LtQ[m,0]
```

```
Int[(f_.*x_)^m_*(d_.+e_.*x_^2)^q_/(a_+c_.*x_^4),x_Symbol] :=
  d/a*Int[(f*x)^m*(d+e*x^2)^(q-1),x] +
  1/(a*f^2)*Int[(f*x)^(m+2)*(d+e*x^2)^(q-1)*Simp[a*e-c*d*x^2,x]/(a+c*x^4),x] /;
FreeQ[{a,c,d,e,f},x] && Not[IntegerQ[q]] && GtQ[q,0] && LtQ[m,0]
```

2. $\int \frac{(f\,x)^m\,(d+e\,x^2)^q}{a+b\,x^2+c\,x^4}\,dx$ when $b^2-4\,a\,c\neq 0\,\wedge\,q\notin\mathbb{Z}\,\wedge\,q<-1$

1. $\int \frac{(f\,x)^m\,(d+e\,x^2)^q}{a+b\,x^2+c\,x^4}\,dx$ when $b^2-4\,a\,c\neq 0\,\wedge\,q\notin\mathbb{Z}\,\wedge\,q<-1\,\wedge\,m>1$

1: $\int \frac{(f\,x)^m\,(d+e\,x^2)^q}{a+b\,x^2+c\,x^4}\,dx$ when $b^2-4\,a\,c\neq 0\,\wedge\,q\notin\mathbb{Z}\,\wedge\,q<-1\,\wedge\,m>3$

Reference: Algebraic expansion

Basis: $\frac{1}{a+b\,z+c\,z^2}==\frac{d^2}{(c\,d^2-b\,d\,e+a\,e^2)\,z^2}-\frac{(d+e\,z)\,(a\,d+(b\,d-a\,e)\,z)}{(c\,d^2-b\,d\,e+a\,e^2)\,z^2\,(a+b\,z+c\,z^2)}$

Rule 1.2.2.4.9.2.2.1.1: If $b^2-4\,a\,c\neq 0\,\wedge\,q\notin\mathbb{Z}\,\wedge\,q<-1\,\wedge\,m>3$, then

$$\int \frac{(f\,x)^m\,(d+e\,x^2)^q}{a+b\,x^2+c\,x^4}\,dx \to \frac{d^2\,f^4}{c\,d^2-b\,d\,e+a\,e^2}\int (f\,x)^{m-4}\,(d+e\,x^2)^q\,dx-\frac{f^4}{c\,d^2-b\,d\,e+a\,e^2}\int \frac{(f\,x)^{m-4}\,(d+e\,x^2)^{q+1}\,(a\,d+(b\,d-a\,e)\,x^2)}{a+b\,x^2+c\,x^4}\,dx$$

–

Program code:

```
Int[(f_.*x_)^m_.*(d_.+e_.*x_^2)^q_/(a_+b_.*x_^2+c_.*x_^4),x_Symbol] :=
  d^2*f^4/(c*d^2-b*d*e+a*e^2)*Int[(f*x)^(m-4)*(d+e*x^2)^q,x] -
  f^4/(c*d^2-b*d*e+a*e^2)*Int[(f*x)^(m-4)*(d+e*x^2)^(q+1)*Simp[a*d+(b*d-a*e)*x^2,x]/(a+b*x^2+c*x^4),x] /;
FreeQ[{a,b,c,d,e,f},x] && NeQ[b^2-4*a*c,0] && Not[IntegerQ[q]] && LtQ[q,-1] && GtQ[m,3]
```

```
Int[(f_.*x_)^m_.*(d_.+e_.*x_^2)^q_/(a_+c_.*x_^4),x_Symbol] :=
  d^2*f^4/(c*d^2+a*e^2)*Int[(f*x)^(m-4)*(d+e*x^2)^q,x] -
  a*f^4/(c*d^2+a*e^2)*Int[(f*x)^(m-4)*(d+e*x^2)^(q+1)*(d-e*x^2)/(a+c*x^4),x] /;
FreeQ[{a,c,d,e,f},x] && Not[IntegerQ[q]] && LtQ[q,-1] && GtQ[m,3]
```

2: $\int \frac{(f x)^m (d+e x^2)^q}{a+b x^2+c x^4} \, dx$ when $b^2-4 a c \neq 0 \wedge q \notin \mathbb{Z} \wedge q<-1 \wedge 1<m \leq 3$

Reference: Algebraic expansion

Basis: $\frac{1}{a+b z+c z^2} == -\frac{d e}{(c d^2-b d e+a e^2) z}+\frac{(d+e z) (a e+c d z)}{(c d^2-b d e+a e^2) z (a+b z+c z^2)}$

Rule 1.2.2.4.9.2.2.1.2: If $b^2-4 a c \neq 0 \wedge q \notin \mathbb{Z} \wedge q<-1 \wedge 1<m \leq 3$, then

$$\int \frac{(f x)^m (d+e x^2)^q}{a+b x^2+c x^4} \, dx \rightarrow -\frac{d e f^2}{c d^2-b d e+a e^2} \int (f x)^{m-2} (d+e x^2)^q \, dx+\frac{f^2}{c d^2-b d e+a e^2} \int \frac{(f x)^{m-2} (d+e x^2)^{q+1} (a e+c d x^2)}{a+b x^2+c x^4} \, dx$$

Program code:

```
Int[(f_.*x_)^m_.*(d_.+e_.*x_^2)^q_/(a_+b_.*x_^2+c_.*x_^4),x_Symbol] :=
  -d*e*f^2/(c*d^2-b*d*e+a*e^2)*Int[(f*x)^(m-2)*(d+e*x^2)^q,x] +
  f^2/(c*d^2-b*d*e+a*e^2)*Int[(f*x)^(m-2)*(d+e*x^2)^(q+1)*Simp[a*e+c*d*x^2,x]/(a+b*x^2+c*x^4),x] /;
FreeQ[{a,b,c,d,e,f},x] && NeQ[b^2-4*a*c,0] && Not[IntegerQ[q]] && LtQ[q,-1] && GtQ[m,1] && LeQ[m,3]
```

```
Int[(f_.*x_)^m_.*(d_.+e_.*x_^2)^q_/(a_+c_.*x_^4),x_Symbol] :=
  -d*e*f^2/(c*d^2+a*e^2)*Int[(f*x)^(m-2)*(d+e*x^2)^q,x] +
  f^2/(c*d^2+a*e^2)*Int[(f*x)^(m-2)*(d+e*x^2)^(q+1)*Simp[a*e+c*d*x^2,x]/(a+c*x^4),x] /;
FreeQ[{a,c,d,e,f},x] && Not[IntegerQ[q]] && LtQ[q,-1] && GtQ[m,1] && LeQ[m,3]
```

2: $\int \frac{(f x)^m (d+e x^2)^q}{a+b x^2+c x^4} \, dx$ when $b^2-4 a c \neq 0 \wedge q \notin \mathbb{Z} \wedge q<-1$

Derivation: Algebraic expansion

Basis: $\frac{1}{a+b z+c z^2} == \frac{e^2}{c d^2-b d e+a e^2}+\frac{(d+e z) (c d-b e-c e z)}{(c d^2-b d e+a e^2) (a+b z+c z^2)}$

Rule 1.2.2.4.9.2.2.2: If $b^2-4 a c \neq 0 \wedge q \notin \mathbb{Z} \wedge q<-1$, then

$$\int \frac{(f x)^m \left(d+e x^2\right)^q}{a+b x^2+c x^4}\,dx \to \frac{e^2}{c d^2-b d e+a e^2}\int (f x)^m \left(d+e x^2\right)^q dx+\frac{1}{c d^2-b d e+a e^2}\int \frac{(f x)^m \left(d+e x^2\right)^{q+1} \left(c d-b e-c e x^2\right)}{a+b x^2+c x^4}\,dx$$

Program code:

```
Int[(f_.*x_)^m_.*(d_+e_.*x_^2)^q_/(a_+b_.*x_^2+c_.*x_^4),x_Symbol] :=
  e^2/(c*d^2-b*d*e+a*e^2)*Int[(f*x)^m*(d+e*x^2)^q,x] +
  1/(c*d^2-b*d*e+a*e^2)*Int[(f*x)^m*(d+e*x^2)^(q+1)*Simp[c*d-b*e-c*e*x^2,x]/(a+b*x^2+c*x^4),x] /;
FreeQ[{a,b,c,d,e,f,m},x] && NeQ[b^2-4*a*c,0] && Not[IntegerQ[q]] && LtQ[q,-1]
```

```
Int[(f_.*x_)^m_.*(d_+e_.*x_^2)^q_/(a_+c_.*x_^4),x_Symbol] :=
  e^2/(c*d^2+a*e^2)*Int[(f*x)^m*(d+e*x^2)^q,x] +
  c/(c*d^2+a*e^2)*Int[(f*x)^m*(d+e*x^2)^(q+1)*(d-e*x^2)/(a+c*x^4),x] /;
FreeQ[{a,c,d,e,f,m},x] && Not[IntegerQ[q]] && LtQ[q,-1]
```

3: $\int \frac{(f x)^m \left(d+e x^2\right)^q}{a+b x^2+c x^4}\,dx$ when $b^2-4 a c\neq 0 \land q\notin \mathbb{Z} \land m\in \mathbb{Z}$

Derivation: Algebraic expansion

- Basis: If $q=\sqrt{b^2-4 a c}$, then $\frac{1}{a+b z+c z^2}=\frac{2 c}{q\,(b-q+2 c z)}-\frac{2 c}{q\,(b+q+2 c z)}$

- Rule 1.2.2.4.9.2.3: If $b^2-4 a c\neq 0 \land q\notin \mathbb{Z} \land m\in \mathbb{Z}$, then

$$\int \frac{(f x)^m \left(d+e x^2\right)^q}{a+b x^2+c x^4}\,dx \to \int \left(d+e x^2\right)^q \text{ExpandIntegrand}\left[\frac{(f x)^m}{a+b x^2+c x^4},\,x\right]dx$$

Program code:

```
Int[(f_.*x_)^m_.*(d_+e_.*x_^2)^q_/(a_+b_.*x_^2+c_.*x_^4),x_Symbol] :=
  Int[ExpandIntegrand[(d+e*x^2)^q,(f*x)^m/(a+b*x^2+c*x^4),x],x] /;
FreeQ[{a,b,c,d,e,f,q},x] && NeQ[b^2-4*a*c,0] && Not[IntegerQ[q]] && IntegerQ[m]
```

```
Int[(f_.*x_)^m_.*(d_+e_.*x_^2)^q_/(a_+c_.*x_^4),x_Symbol] :=
  Int[ExpandIntegrand[(d+e*x^2)^q,(f*x)^m/(a+c*x^4),x],x] /;
FreeQ[{a,c,d,e,f,q},x] && Not[IntegerQ[q]] && IntegerQ[m]
```

4: $\int \frac{(f x)^m \left(d+e x^2\right)^q}{a+b x^2+c x^4} \, dx$ when $b^2-4 a c \neq 0 \land q \notin \mathbb{Z} \land m \notin \mathbb{Z}$

Derivation: Algebraic expansion

- Basis: If $q=\sqrt{b^2-4 a c}$, then $\frac{1}{a+b z+c z^2}==\frac{2 c}{q (b-q+2 c z)}-\frac{2 c}{q (b+q+2 c z)}$

- Rule 1.2.2.4.9.2.4: If $b^2-4 a c \neq 0 \land q \notin \mathbb{Z} \land m \notin \mathbb{Z}$, then

$$\int \frac{(f x)^m \left(d+e x^2\right)^q}{a+b x^2+c x^4} \, dx \to \int (f x)^m \left(d+e x^2\right)^q \text{ExpandIntegrand}\left[\frac{1}{a+b x^2+c x^4}, x\right] dx$$

Program code:

```
Int[(f_.*x_)^m_.*(d_+e_.*x_^2)^q_/(a_+b_.*x_^2+c_.*x_^4),x_Symbol] :=
  Int[ExpandIntegrand[(f*x)^m*(d+e*x^2)^q,1/(a+b*x^2+c*x^4),x],x] /;
FreeQ[{a,b,c,d,e,f,m,q},x] && NeQ[b^2-4*a*c,0] && Not[IntegerQ[q]] && Not[IntegerQ[m]]
```

```
Int[(f_.*x_)^m_.*(d_+e_.*x_^2)^q_/(a_+c_.*x_^4),x_Symbol] :=
  Int[ExpandIntegrand[(f*x)^m*(d+e*x^2)^q,1/(a+c*x^4),x],x] /;
FreeQ[{a,c,d,e,f,m,q},x] && Not[IntegerQ[q]] && Not[IntegerQ[m]]
```

10: $\int \frac{(f\,x)^m\,(d+e\,x^2)^q}{a+b\,x^2+c\,x^4}\,dx$ when $b^2-4\,a\,c\neq 0$

Derivation: Algebraic expansion

Basis: If $r=\sqrt{b^2-4\,a\,c}$, then $\frac{1}{a+b\,z+c\,z^2}=\frac{2\,c}{r\,(b-r+2\,c\,z)}-\frac{2\,c}{r\,(b+r+2\,c\,z)}$

Rule 1.2.2.4.10: If $b^2-4\,a\,c\neq 0$, then

$$\int \frac{(f\,x)^m\,(d+e\,x^2)^q}{a+b\,x^2+c\,x^4}\,dx\;\to\;\frac{2\,c}{r}\int \frac{(f\,x)^m\,(d+e\,x^2)^q}{b-r+2\,c\,x^2}\,dx-\frac{2\,c}{r}\int \frac{(f\,x)^m\,(d+e\,x^2)^q}{b+r+2\,c\,x^2}\,dx$$

Program code:

```
Int[(f_.*x_)^m_.*(d_+e_.*x_^2)^q_/(a_+b_.*x_^2+c_.*x_^4),x_Symbol] :=
  With[{r=Rt[b^2-4*a*c,2]},
  2*c/r*Int[(f*x)^m*(d+e*x^2)^q/(b-r+2*c*x^2),x] - 2*c/r*Int[(f*x)^m*(d+e*x^2)^q/(b+r+2*c*x^2),x]] /;
FreeQ[{a,b,c,d,e,f,m,q},x] && NeQ[b^2-4*a*c,0]
```

```
Int[(f_.*x_)^m_.*(d_+e_.*x_^2)^q_/(a_+c_.*x_^4),x_Symbol] :=
  With[{r=Rt[-a*c,2]},
  -c/(2*r)*Int[(f*x)^m*(d+e*x^2)^q/(r-c*x^2),x] - c/(2*r)*Int[(f*x)^m*(d+e*x^2)^q/(r+c*x^2),x]] /;
FreeQ[{a,c,d,e,f,m,q},x]
```

11. $\int \frac{(f\,x)^m\,\left(a+b\,x^2+c\,x^4\right)^p}{d+e\,x^2}\,dx$ when $b^2-4\,a\,c\neq 0$

1. $\int \frac{(f\,x)^m\,\left(a+b\,x^2+c\,x^4\right)^p}{d+e\,x^2}\,dx$ when $b^2-4\,a\,c\neq 0\,\land\,p>0\,\land\,m<0$

1: $\int \frac{(f\,x)^m\,\left(a+b\,x^2+c\,x^4\right)^p}{d+e\,x^2}\,dx$ when $b^2-4\,a\,c\neq 0\,\land\,p>0\,\land\,m<-2$

Reference: Algebraic expansion

Basis: $\frac{a+b\,z+c\,z^2}{d+e\,z}==\frac{a\,d+(b\,d-a\,e)\,z}{d^2}+\frac{\left(c\,d^2-b\,d\,e+a\,e^2\right)\,z^2}{d^2\,(d+e\,z)}$

Rule 1.2.2.4.11.1.1: If $b^2-4\,a\,c\neq 0\,\land\,p>0\,\land\,m<-2$, then

$$\int \frac{(f\,x)^m\,\left(a+b\,x^2+c\,x^4\right)^p}{d+e\,x^2}\,dx\,\longrightarrow$$

$$\frac{1}{d^2}\int (f\,x)^m\,\left(a\,d+(b\,d-a\,e)\,x^2\right)\,\left(a+b\,x^2+c\,x^4\right)^{p-1}\,dx+\frac{c\,d^2-b\,d\,e+a\,e^2}{d^2\,f^4}\int \frac{(f\,x)^{m+4}\,\left(a+b\,x^2+c\,x^4\right)^{p-1}}{d+e\,x^2}\,dx$$

Program code:

```
Int[(f_.*x_)^m_*(a_.+b_.*x_^2+c_.*x_^4)^p_./(d_.+e_.*x_^2),x_Symbol] :=
  1/d^2*Int[(f*x)^m*(a*d+(b*d-a*e)*x^2)*(a+b*x^2+c*x^4)^(p-1),x] +
  (c*d^2-b*d*e+a*e^2)/(d^2*f^4)*Int[(f*x)^(m+4)*(a+b*x^2+c*x^4)^(p-1)/(d+e*x^2),x] /;
FreeQ[{a,b,c,d,e,f},x] && NeQ[b^2-4*a*c,0] && GtQ[p,0] && LtQ[m,-2]
```

```
Int[(f_.*x_)^m_*(a_+c_.*x_^4)^p_./(d_.+e_.*x_^2),x_Symbol] :=
  a/d^2*Int[(f*x)^m*(d-e*x^2)*(a+c*x^4)^(p-1),x] +
  (c*d^2+a*e^2)/(d^2*f^4)*Int[(f*x)^(m+4)*(a+c*x^4)^(p-1)/(d+e*x^2),x] /;
FreeQ[{a,c,d,e,f},x] && GtQ[p,0] && LtQ[m,-2]
```

2: $\int \frac{(f x)^m \left(a+b x^2+c x^4\right)^p}{d+e x^2} \, dx$ when $b^2-4 a c \neq 0 \wedge p>0 \wedge m<0$

Reference: Algebraic expansion

Basis: $\frac{a+b z+c z^2}{d+e z} = \frac{a e+c d z}{d e} - \frac{\left(c d^2-b d e+a e^2\right) z}{d e (d+e z)}$

Rule 1.2.2.4.11.1.2: If $b^2-4 a c \neq 0 \wedge p>0 \wedge m<0$, then

$$\int \frac{(f x)^m \left(a+b x^2+c x^4\right)^p}{d+e x^2} \, dx \longrightarrow$$

$$\frac{1}{d e} \int (f x)^m \left(a e+c d x^2\right) \left(a+b x^2+c x^4\right)^{p-1} dx - \frac{c d^2-b d e+a e^2}{d e f^2} \int \frac{(f x)^{m+2} \left(a+b x^2+c x^4\right)^{p-1}}{d+e x^2} \, dx$$

Program code:

```
Int[(f_.*x_)^m_*(a_.+b_.*x_^2+c_.*x_^4)^p_./(d_.+e_.*x_^2),x_Symbol] :=
  1/(d*e)*Int[(f*x)^m*(a*e+c*d*x^2)*(a+b*x^2+c*x^4)^(p-1),x] -
  (c*d^2-b*d*e+a*e^2)/(d*e*f^2)*Int[(f*x)^(m+2)*(a+b*x^2+c*x^4)^(p-1)/(d+e*x^2),x] /;
FreeQ[{a,b,c,d,e,f},x] && NeQ[b^2-4*a*c,0] && GtQ[p,0] && LtQ[m,0]
```

```
Int[(f_.*x_)^m_*(a_+c_.*x_^4)^p_./(d_.+e_.*x_^2),x_Symbol] :=
  1/(d*e)*Int[(f*x)^m*(a*e+c*d*x^2)*(a+c*x^4)^(p-1),x] -
  (c*d^2+a*e^2)/(d*e*f^2)*Int[(f*x)^(m+2)*(a+c*x^4)^(p-1)/(d+e*x^2),x] /;
FreeQ[{a,c,d,e,f},x] && GtQ[p,0] && LtQ[m,0]
```

2. $\int \frac{(f x)^m \left(a+b x^2+c x^4\right)^p}{d+e x^2} \, dx$ when $b^2-4 a c \neq 0 \land p<-1 \land m>0$

1: $\int \frac{(f x)^m \left(a+b x^2+c x^4\right)^p}{d+e x^2} \, dx$ when $b^2-4 a c \neq 0 \land p<-1 \land m>2$

Reference: Algebraic expansion

Basis: $\frac{z^2}{d+e z} = -\frac{a d+(b d-a e) z}{c d^2-b d e+a e^2}+\frac{d^2 \left(a+b z+c z^2\right)}{\left(c d^2-b d e+a e^2\right) (d+e z)}$

Rule 1.2.2.4.11.2.1: If $b^2-4 a c \neq 0 \land p<-1 \land m>2$, then

$$\int \frac{(f x)^m \left(a+b x^2+c x^4\right)^p}{d+e x^2} \, dx \longrightarrow$$

$$-\frac{f^4}{c d^2-b d e+a e^2} \int (f x)^{m-4} \left(a d+(b d-a e) x^2\right) \left(a+b x^2+c x^4\right)^p \, dx+\frac{d^2 f^4}{c d^2-b d e+a e^2} \int \frac{(f x)^{m-4} \left(a+b x^2+c x^4\right)^{p+1}}{d+e x^2} \, dx$$

Program code:

```
Int[(f_.*x_)^m_.*(a_.+b_.*x_^2+c_.*x_^4)^p_/(d_.+e_.*x_^2),x_Symbol] :=
  -f^4/(c*d^2-b*d*e+a*e^2)*Int[(f*x)^(m-4)*(a*d+(b*d-a*e)*x^2)*(a+b*x^2+c*x^4)^p,x] +
  d^2*f^4/(c*d^2-b*d*e+a*e^2)*Int[(f*x)^(m-4)*(a+b*x^2+c*x^4)^(p+1)/(d+e*x^2),x] /;
FreeQ[{a,b,c,d,e,f},x] && NeQ[b^2-4*a*c,0] && LtQ[p,-1] && GtQ[m,2]
```

```
Int[(f_.*x_)^m_.*(a_+c_.*x_^4)^p_/(d_.+e_.*x_^2),x_Symbol] :=
  -a*f^4/(c*d^2+a*e^2)*Int[(f*x)^(m-4)*(d-e*x^2)*(a+c*x^4)^p,x] +
  d^2*f^4/(c*d^2+a*e^2)*Int[(f*x)^(m-4)*(a+c*x^4)^(p+1)/(d+e*x^2),x] /;
FreeQ[{a,c,d,e,f},x] && LtQ[p,-1] && GtQ[m,2]
```

2: $\int \frac{(f x)^m \left(a+b x^2+c x^4\right)^p}{d+e x^2} \, dx$ when $b^2-4 a c \neq 0 \land p<-1 \land m>0$

Reference: Algebraic expansion

Basis: $\frac{z}{d+e z} = \frac{a e+c d z}{c d^2-b d e+a e^2}-\frac{d e \left(a+b z+c z^2\right)}{\left(c d^2-b d e+a e^2\right) (d+e z)}$

–

Rule 1.2.2.4.11.2.2: If $b^2-4 a c \neq 0 \land p<-1 \land m>0$, then

$$\int \frac{(f x)^m \left(a+b x^2+c x^4\right)^p}{d+e x^2} \, dx \to$$

$$\frac{f^2}{c d^2-b d e+a e^2} \int (f x)^{m-2} \left(a e+c d x^2\right) \left(a+b x^2+c x^4\right)^p \, dx-\frac{d e f^2}{c d^2-b d e+a e^2} \int \frac{(f x)^{m-2} \left(a+b x^2+c x^4\right)^{p+1}}{d+e x^2} \, dx$$

Program code:

```
Int[(f_.*x_)^m_.*(a_.+b_.*x_^2+c_.*x_^4)^p_/(d_.+e_.*x_^2),x_Symbol] :=
  f^2/(c*d^2-b*d*e+a*e^2)*Int[(f*x)^(m-2)*(a*e+c*d*x^2)*(a+b*x^2+c*x^4)^p,x] -
  d*e*f^2/(c*d^2-b*d*e+a*e^2)*Int[(f*x)^(m-2)*(a+b*x^2+c*x^4)^(p+1)/(d+e*x^2),x] /;
FreeQ[{a,b,c,d,e,f},x] && NeQ[b^2-4*a*c,0] && LtQ[p,-1] && GtQ[m,0]
```

```
Int[(f_.*x_)^m_.*(a_+c_.*x_^4)^p_/(d_.+e_.*x_^2),x_Symbol] :=
  f^2/(c*d^2+a*e^2)*Int[(f*x)^(m-2)*(a*e+c*d*x^2)*(a+c*x^4)^p,x] -
  d*e*f^2/(c*d^2+a*e^2)*Int[(f*x)^(m-2)*(a+c*x^4)^(p+1)/(d+e*x^2),x] /;
FreeQ[{a,c,d,e,f},x] && LtQ[p,-1] && GtQ[m,0]
```

3. $\int \frac{x^m}{\left(d+e x^2\right) \sqrt{a+b x^2+c x^4}} \, dx$ when $b^2-4 a c \neq 0 \land \frac{m}{2} \in \mathbb{Z}$

1. $\int \frac{x^m}{\left(d+e x^2\right) \sqrt{a+b x^2+c x^4}} \, dx$ when $b^2-4 a c \neq 0 \land \frac{m}{2} \in \mathbb{Z}^+$

1. $\int \frac{x^2}{(d+e x^2)\sqrt{a+b x^2+c x^4}}\,dx$ when $b^2-4ac\neq 0$

1: $\int \frac{x^2}{(d+e x^2)\sqrt{a+b x^2+c x^4}}\,dx$ when $b^2-4ac\neq 0 \land c d^2-b d e+a e^2\neq 0 \land \frac{c}{a}>0 \land c d^2-a e^2==0$

Derivation: Algebraic expansion

Rule 1.2.2.4.11.3.1.1.1: If $b^2-4ac\neq 0 \land c d^2-b d e+a e^2\neq 0 \land \frac{c}{a}>0 \land c d^2-a e^2==0$, then

$$\int \frac{x^2}{(d+e x^2)\sqrt{a+b x^2+c x^4}}\,dx \to \frac{d}{2 d e}\int \frac{1}{\sqrt{a+b x^2+c x^4}}\,dx-\frac{d}{2 d e}\int \frac{d-e x^2}{(d+e x^2)\sqrt{a+b x^2+c x^4}}\,dx$$

Program code:

```
Int[x_^2/((d_+e_.*x_^2)*Sqrt[a_+b_.*x_^2+c_.*x_^4]),x_Symbol] :=
  d/(2*d*e)*Int[1/Sqrt[a+b*x^2+c*x^4],x] -
  d/(2*d*e)*Int[(d-e*x^2)/((d+e*x^2)*Sqrt[a+b*x^2+c*x^4]),x] /;
FreeQ[{a,b,c,d,e},x] && NeQ[b^2-4*a*c,0] && NeQ[c*d^2-b*d*e+a*e^2,0] && PosQ[c/a] && EqQ[c*d^2-a*e^2,0]
```

```
Int[x_^2/((d_+e_.*x_^2)*Sqrt[a_+c_.*x_^4]),x_Symbol] :=
  d/(2*d*e)*Int[1/Sqrt[a+c*x^4],x] -
  d/(2*d*e)*Int[(d-e*x^2)/((d+e*x^2)*Sqrt[a+c*x^4]),x] /;
FreeQ[{a,c,d,e},x] && NeQ[c*d^2+a*e^2,0] && PosQ[c/a] && EqQ[c*d^2-a*e^2,0]
```

2: $\int \frac{x^2}{(d+e x^2)\sqrt{a+b x^2+c x^4}}\,dx$ when $b^2-4ac\neq 0 \land c d^2-b d e+a e^2\neq 0 \land \frac{c}{a}>0 \land c d^2-a e^2\neq 0$

Derivation: Algebraic expansion

Basis: $\frac{x^2}{d+e x^2}==\frac{1}{e-d q}-\frac{d\,(1+q x^2)}{(e-d q)\,(d+e x^2)}$

-

Rule 1.2.2.4.11.3.1.1.2: If $b^2-4ac\neq 0 \land c d^2-b d e+a e^2\neq 0 \land \frac{c}{a}>0 \land c d^2-a e^2\neq 0$, let $q\to\sqrt{\frac{c}{a}}$, then

$$\int \frac{x^2}{\left(d+e x^2\right) \sqrt{a+b x^2+c x^4}} \, dx \longrightarrow -\frac{a\,(e+d\,q)}{c\,d^2-a\,e^2} \int \frac{1}{\sqrt{a+b x^2+c x^4}} \, dx + \frac{a\,d\,(e+d\,q)}{c\,d^2-a\,e^2} \int \frac{1+q\,x^2}{\left(d+e x^2\right) \sqrt{a+b x^2+c x^4}} \, dx$$

–

Program code:

```
Int[x_^2/((d_+e_.*x_^2)*Sqrt[a_+b_.*x_^2+c_.*x_^4]),x_Symbol] :=
  With[{q=Rt[c/a,2]},
  -a*(e+d*q)/(c*d^2-a*e^2)*Int[1/Sqrt[a+b*x^2+c*x^4],x] +
  a*d*(e+d*q)/(c*d^2-a*e^2)*Int[(1+q*x^2)/((d+e*x^2)*Sqrt[a+b*x^2+c*x^4]),x]] /;
FreeQ[{a,b,c,d,e},x] && NeQ[b^2-4*a*c,0] && NeQ[c*d^2-b*d*e+a*e^2,0] && PosQ[c/a] && NeQ[c*d^2-a*e^2,0]
```

```
Int[x_^2/((d_+e_.*x_^2)*Sqrt[a_+c_.*x_^4]),x_Symbol] :=
  With[{q=Rt[c/a,2]},
  -a*(e+d*q)/(c*d^2-a*e^2)*Int[1/Sqrt[a+c*x^4],x] +
  a*d*(e+d*q)/(c*d^2-a*e^2)*Int[(1+q*x^2)/((d+e*x^2)*Sqrt[a+c*x^4]),x]] /;
FreeQ[{a,c,d,e},x] && NeQ[c*d^2+a*e^2,0] && PosQ[c/a] && NeQ[c*d^2-a*e^2,0]
```

2. $\int \frac{x^4}{(d+e x^2)\sqrt{a+b x^2+c x^4}} dx$ when $b^2-4 a c \neq 0 \land \frac{c}{a}>0$

1: $\int \frac{x^4}{(d+e x^2)\sqrt{a+b x^2+c x^4}} dx$ when $b^2-4 a c \neq 0 \land \frac{c}{a}>0 \land c d^2-a e^2 == 0$

Derivation: Algebraic expansion

Rule 1.2.2.4.11.3.1.2.1: If $b^2-4 a c \neq 0 \land \frac{c}{a}>0 \land c d^2-a e^2 == 0$, then

$$\int \frac{x^4}{(d+e x^2)\sqrt{a+b x^2+c x^4}} dx \longrightarrow -\frac{1}{e^2}\int \frac{d-e x^2}{\sqrt{a+b x^2+c x^4}} dx+\frac{d^2}{e^2}\int \frac{1}{(d+e x^2)\sqrt{a+b x^2+c x^4}} dx$$

Program code:

```
Int[x_^4/((d_+e_.*x_^2)*Sqrt[a_+b_.*x_^2+c_.*x_^4]),x_Symbol] :=
  -1/e^2*Int[(d-e*x^2)/Sqrt[a+b*x^2+c*x^4],x] + d^2/e^2*Int[1/((d+e*x^2)*Sqrt[a+b*x^2+c*x^4]),x] /;
FreeQ[{a,b,c,d,e},x] && NeQ[b^2-4*a*c,0] && PosQ[c/a] && EqQ[c*d^2-a*e^2,0]
```

```
Int[x_^4/((d_+e_.*x_^2)*Sqrt[a_+c_.*x_^4]),x_Symbol] :=
  -1/e^2*Int[(d-e*x^2)/Sqrt[a+c*x^4],x] + d^2/e^2*Int[1/((d+e*x^2)*Sqrt[a+c*x^4]),x] /;
FreeQ[{a,c,d,e},x] && PosQ[c/a] && EqQ[c*d^2-a*e^2,0]
```

2: $\int \frac{x^4}{(d+e x^2)\sqrt{a+b x^2+c x^4}} dx$ when $b^2-4 a c \neq 0 \land \frac{c}{a}>0 \land c d^2-a e^2 \neq 0$

Derivation: Algebraic expansion

Rule 1.2.2.4.11.3.1.2.2: If $b^2-4 a c \neq 0 \land \frac{c}{a}>0 \land c d^2-a e^2 \neq 0$, let $q \to \sqrt{\frac{c}{a}}$, then

$$\int \frac{x^4}{(d+e x^2)\sqrt{a+b x^2+c x^4}} dx \longrightarrow$$

$$-\frac{2\,c\,d-a\,e\,q}{c\,e\,(e-d\,q)}\int\frac{1}{\sqrt{a+b\,x^2+c\,x^4}}\,dx-\frac{1}{e\,q}\int\frac{1-q\,x^2}{\sqrt{a+b\,x^2+c\,x^4}}\,dx+\frac{d^2}{e\,(e-d\,q)}\int\frac{1+q\,x^2}{\left(d+e\,x^2\right)\sqrt{a+b\,x^2+c\,x^4}}\,dx$$

–

Program code:

```
Int[x_^4/((d_+e_.*x_^2)*Sqrt[a_+b_.*x_^2+c_.*x_^4]),x_Symbol] :=
  With[{q=Rt[c/a,2]},
  -1/(e*q)*Int[(1-q*x^2)/Sqrt[a+b*x^2+c*x^4],x] +
  d^2/(e*(e-d*q))*Int[(1+q*x^2)/((d+e*x^2)*Sqrt[a+b*x^2+c*x^4]),x] /;
 EqQ[2*c*d-a*e*q,0]] /;
FreeQ[{a,b,c,d,e},x] && NeQ[b^2-4*a*c,0] && PosQ[c/a] && NeQ[c*d^2-a*e^2,0]
```

```
Int[x_^4/((d_+e_.*x_^2)*Sqrt[a_+c_.*x_^4]),x_Symbol] :=
  With[{q=Rt[c/a,2]},
  -1/(e*q)*Int[(1-q*x^2)/Sqrt[a+c*x^4],x] +
  d^2/(e*(e-d*q))*Int[(1+q*x^2)/((d+e*x^2)*Sqrt[a+c*x^4]),x] /;
 EqQ[2*c*d-a*e*q,0]] /;
FreeQ[{a,c,d,e},x] && PosQ[c/a] && NeQ[c*d^2-a*e^2,0]
```

```
Int[x_^4/((d_+e_.*x_^2)*Sqrt[a_+b_.*x_^2+c_.*x_^4]),x_Symbol] :=
  With[{q=Rt[c/a,2]},
  -(2*c*d-a*e*q)/(c*e*(e-d*q))*Int[1/Sqrt[a+b*x^2+c*x^4],x] -
  1/(e*q)*Int[(1-q*x^2)/Sqrt[a+b*x^2+c*x^4],x] +
  d^2/(e*(e-d*q))*Int[(1+q*x^2)/((d+e*x^2)*Sqrt[a+b*x^2+c*x^4]),x]] /;
FreeQ[{a,b,c,d,e},x] && NeQ[b^2-4*a*c,0] && PosQ[c/a] && NeQ[c*d^2-a*e^2,0]
```

```
Int[x_^4/((d_+e_.*x_^2)*Sqrt[a_+c_.*x_^4]),x_Symbol] :=
  With[{q=Rt[c/a,2]},
  -(2*c*d-a*e*q)/(c*e*(e-d*q))*Int[1/Sqrt[a+c*x^4],x] -
  1/(e*q)*Int[(1-q*x^2)/Sqrt[a+c*x^4],x] +
  d^2/(e*(e-d*q))*Int[(1+q*x^2)/((d+e*x^2)*Sqrt[a+c*x^4]),x]] /;
FreeQ[{a,c,d,e},x] && PosQ[c/a] && NeQ[c*d^2-a*e^2,0]
```

3: $\int \frac{x^m}{(d+e x^2)\sqrt{a+b x^2+c x^4}}\,dx$ when $b^2-4ac\neq 0 \land \frac{m}{2}-2\in\mathbb{Z}^+$

Rule 1.2.2.4.11.3.1.3: If $b^2-4ac\neq 0 \land \frac{m}{2}-2\in\mathbb{Z}^+$, then

$$\int \frac{x^m}{(d+e x^2)\sqrt{a+b x^2+c x^4}}\,dx \longrightarrow$$

$$\frac{x^{m-5}\sqrt{a+b x^2+c x^4}}{c e (m-3)}-\frac{1}{c e (m-3)}\int \frac{x^{m-6}\left(a d (m-5)+(a e (m-5)+b d (m-4)) x^2+(b e (m-4)+c d (m-3)) x^4\right)}{(d+e x^2)\sqrt{a+b x^2+c x^4}}\,dx$$

Program code:

```
Int[x_^m_/((d_+e_.*x_^2)*Sqrt[a_+b_.*x_^2+c_.*x_^4]),x_Symbol] :=
  x^(m-5)*Sqrt[a+b*x^2+c*x^4]/(c*e*(m-3)) -
  1/(c*e*(m-3))*Int[x^(m-6)/((d+e*x^2)*Sqrt[a+b*x^2+c*x^4])*
    Simp[a*d*(m-5)+(a*e*(m-5)+b*d*(m-4))*x^2+(b*e*(m-4)+c*d*(m-3))*x^4,x],x] /;
FreeQ[{a,b,c,d,e},x] && NeQ[b^2-4*a*c,0] && IGtQ[m/2,2]
```

```
Int[x_^m_/((d_+e_.*x_^2)*Sqrt[a_+c_.*x_^4]),x_Symbol] :=
  x^(m-5)*Sqrt[a+c*x^4]/(c*e*(m-3)) -
  1/(c*e*(m-3))*Int[x^(m-6)/((d+e*x^2)*Sqrt[a+c*x^4])*Simp[a*d*(m-5)+a*e*(m-5)*x^2+c*d*(m-3)*x^4,x],x] /;
FreeQ[{a,c,d,e},x] && IGtQ[m/2,2]
```

2: $\int \frac{x^m}{(d+e x^2)\sqrt{a+b x^2+c x^4}}\,dx$ when $b^2-4ac\neq 0 \land \frac{m}{2}\in\mathbb{Z}^-$

Rule 1.2.2.4.11.3.2: If $b^2-4ac\neq 0 \land \frac{m}{2}\in\mathbb{Z}^-$, then

$$\int \frac{x^m}{(d+e x^2)\sqrt{a+b x^2+c x^4}}\,dx \longrightarrow$$

$$\frac{x^{m+1}\sqrt{a+b x^2+c x^4}}{a d (m+1)}-\frac{1}{a d (m+1)}\int\frac{x^{m+2}\left(a e (m+1)+b d (m+2)+(b e (m+2)+c d (m+3)) x^2+c e (m+3) x^4\right)}{\left(d+e x^2\right)\sqrt{a+b x^2+c x^4}}\,dx$$

Program code:

```
Int[x_^m_/((d_+e_.*x_^2)*Sqrt[a_+b_.*x_^2+c_.*x_^4]),x_Symbol] :=
  x^(m+1)*Sqrt[a+b*x^2+c*x^4]/(a*d*(m+1)) -
  1/(a*d*(m+1))*Int[x^(m+2)/((d+e*x^2)*Sqrt[a+b*x^2+c*x^4])*
    Simp[a*e*(m+1)+b*d*(m+2)+(b*e*(m+2)+c*d*(m+3))*x^2+c*e*(m+3)*x^4,x],x] /;
FreeQ[{a,b,c,d,e},x] && NeQ[b^2-4*a*c,0] && ILtQ[m/2,0]
```

```
Int[x_^m_/((d_+e_.*x_^2)*Sqrt[a_+c_.*x_^4]),x_Symbol] :=
  x^(m+1)*Sqrt[a+c*x^4]/(a*d*(m+1)) -
  1/(a*d*(m+1))*Int[x^(m+2)/((d+e*x^2)*Sqrt[a+c*x^4])*Simp[a*e*(m+1)+c*d*(m+3)*x^2+c*e*(m+3)*x^4,x],x] /;
FreeQ[{a,c,d,e},x] && ILtQ[m/2,0]
```

12: $\int\frac{x^m}{\sqrt{d+e x^2}\sqrt{a+b x^2+c x^4}}\,dx$ when $b^2-4 a c\neq 0\wedge\frac{m}{2}\in\mathbb{Z}$

Derivation: Piecewise constant extraction

Basis: $\partial_x\frac{x\sqrt{e+\frac{d}{x^2}}}{\sqrt{d+e x^2}}==0$

Basis: $\partial_x\frac{x^2\sqrt{c+\frac{b}{x^2}+\frac{a}{x^4}}}{\sqrt{a+b x^2+c x^4}}==0$

Note: Since $m-3$ is odd, the resulting integrand can be reduced to an integrand of the form $\frac{1}{x^{m/2}\sqrt{e+d x}\sqrt{c+b x+a x^2}}$ using the substitution $x\to\frac{1}{x^2}$.

Rule 1.2.2.4.12: If $b^2-4 a c\neq 0\wedge\frac{m}{2}\in\mathbb{Z}$, then

$$\int \frac{x^m}{\sqrt{d+e x^2}\ \sqrt{a+b x^2+c x^4}}\, dx \to \frac{x^3 \sqrt{e+\frac{d}{x^2}}\ \sqrt{c+\frac{b}{x^2}+\frac{a}{x^4}}}{\sqrt{d+e x^2}\ \sqrt{a+b x^2+c x^4}} \int \frac{x^{m-3}}{\sqrt{e+\frac{d}{x^2}}\ \sqrt{c+\frac{b}{x^2}+\frac{a}{x^4}}}\, dx$$

– Program code:

```
Int[x_^m_/(Sqrt[d_+e_.*x_^2]*Sqrt[a_+b_.*x_^2+c_.*x_^4]),x_Symbol] :=
  x^3*Sqrt[e+d/x^2]*Sqrt[c+b/x^2+a/x^4]/(Sqrt[d+e*x^2]*Sqrt[a+b*x^2+c*x^4])*
    Int[x^(m-3)/(Sqrt[e+d/x^2]*Sqrt[c+b/x^2+a/x^4]),x] /;
FreeQ[{a,b,c,d,e},x] && NeQ[b^2-4*a*c,0] && IntegerQ[m/2]
```

```
Int[x_^m_/(Sqrt[d_+e_.*x_^2]*Sqrt[a_+c_.*x_^4]),x_Symbol] :=
  x^3*Sqrt[e+d/x^2]*Sqrt[c+a/x^4]/(Sqrt[d+e*x^2]*Sqrt[a+c*x^4])*
    Int[x^(m-3)/(Sqrt[e+d/x^2]*Sqrt[c+a/x^4]),x] /;
FreeQ[{a,c,d,e},x] && IntegerQ[m/2]
```

13. $\int x^m \left(d+e x^2\right)^q \left(a+b x^2+c x^4\right)^p dx$ when $b^2-4 a c \neq 0 \land p<-1 \land q-1 \in \mathbb{Z}^+ \land \frac{m}{2} \in \mathbb{Z}$

1: $\int x^m \left(d+e x^2\right)^q \left(a+b x^2+c x^4\right)^p dx$ when $b^2-4 a c \neq 0 \land p<-1 \land q-1 \in \mathbb{Z}^+ \land \frac{m}{2} \in \mathbb{Z}^+$

Derivation: Algebraic expansion and trinomial recurrence 2b

– Rule 1.2.2.4.13.1: If $b^2-4 a c \neq 0 \land p<-1 \land q-1 \in \mathbb{Z}^+ \land \frac{m}{2} \in \mathbb{Z}^+$,

let $Q[x] \to \text{PolynomialQuotient}\left[x^m \left(d+e x^2\right)^q, a+b x^2+c x^4, x\right]$ and

$f+g x^2 \to \text{PolynomialRemainder}\left[x^m \left(d+e x^2\right)^q, a+b x^2+c x^4, x\right]$, then

$$\int x^m \left(d+e x^2\right)^q \left(a+b x^2+c x^4\right)^p dx \longrightarrow$$

$$\int \left(f+g x^2\right) \left(a+b x^2+c x^4\right)^p dx+\int Q[x] \left(a+b x^2+c x^4\right)^{p+1} dx \longrightarrow$$

$$\frac{x \left(a+b x^2+c x^4\right)^{p+1} \left(a b g-f \left(b^2-2 a c\right)-c \left(b f-2 a g\right) x^2\right)}{2 a (p+1) \left(b^2-4 a c\right)}+$$

$$\frac{1}{2 a (p+1) \left(b^2-4 a c\right)} \int \left(a+b x^2+c x^4\right)^{p+1} \cdot$$
$$\left(2 a (p+1) \left(b^2-4 a c\right) Q[x]+b^2 f (2 p+3)-2 a c f (4 p+5)-a b g+c (4 p+7) \left(b f-2 a g\right) x^2\right) dx$$

Program code:

```
Int[x_^m_*(d_+e_.*x_^2)^q_*(a_+b_.*x_^2+c_.*x_^4)^p_,x_Symbol] :=
  With[{f=Coeff[PolynomialRemainder[x^m*(d+e*x^2)^q,a+b*x^2+c*x^4,x],x,0],
        g=Coeff[PolynomialRemainder[x^m*(d+e*x^2)^q,a+b*x^2+c*x^4,x],x,2]},
  x*(a+b*x^2+c*x^4)^(p+1)*(a*b*g-f*(b^2-2*a*c)-c*(b*f-2*a*g)*x^2)/(2*a*(p+1)*(b^2-4*a*c)) +
  1/(2*a*(p+1)*(b^2-4*a*c))*Int[(a+b*x^2+c*x^4)^(p+1)*
    Simp[ExpandToSum[2*a*(p+1)*(b^2-4*a*c)*PolynomialQuotient[x^m*(d+e*x^2)^q,a+b*x^2+c*x^4,x]+
      b^2*f*(2*p+3)-2*a*c*f*(4*p+5)-a*b*g+c*(4*p+7)*(b*f-2*a*g)*x^2,x],x],x]] /;
FreeQ[{a,b,c,d,e},x] && NeQ[b^2-4*a*c,0] && LtQ[p,-1] && IGtQ[q,1] && IGtQ[m/2,0]
```

2: $\int x^m \left(d+e x^2\right)^q \left(a+b x^2+c x^4\right)^p dx$ when $b^2-4 a c\neq 0 \land p<-1 \land q-1\in\mathbb{Z}^+ \land \frac{m}{2}\in\mathbb{Z}^-$

Derivation: Algebraic expansion and trinomial recurrence 2b

Rule 1.2.2.4.13.2: If $b^2-4 a c\neq 0 \land p<-1 \land q-1\in\mathbb{Z}^+ \land \frac{m}{2}\in\mathbb{Z}^-$,

let $Q[x]\to \text{PolynomialQuotient}\left[x^m \left(d+e x^2\right)^q,\ a+b x^2+c x^4,\ x\right]$ and $f+g x^2\to \text{PolynomialRemainder}\left[x^m \left(d+e x^2\right)^q,\ a+b x^2+c x^4,\ x\right]$, then

$$\int x^m \left(d+e x^2\right)^q \left(a+b x^2+c x^4\right)^p dx \longrightarrow$$

$$\int \left(f+g x^2\right) \left(a+b x^2+c x^4\right)^p dx+\int Q[x] \left(a+b x^2+c x^4\right)^{p+1} dx \longrightarrow$$

$$\frac{x \left(a+b x^2+c x^4\right)^{p+1} \left(a b g-f \left(b^2-2 a c\right)-c \left(b f-2 a g\right) x^2\right)}{2 a (p+1) \left(b^2-4 a c\right)}+\frac{1}{2 a (p+1) \left(b^2-4 a c\right)}\int x^m \left(a+b x^2+c x^4\right)^{p+1}\cdot$$
$$\left(2 a (p+1) \left(b^2-4 a c\right) x^{-m} Q[x]+\left(b^2 f (2 p+3)-2 a c f (4 p+5)-a b g\right) x^{-m}+c (4 p+7) \left(b f-2 a g\right) x^{2-m}\right) dx$$

Program code:

```
Int[x_^m_*(d_+e_.*x_^2)^q_*(a_+b_.*x_^2+c_.*x_^4)^p_,x_Symbol] :=
  With[{f=Coeff[PolynomialRemainder[x^m*(d+e*x^2)^q,a+b*x^2+c*x^4,x],x,0],
        g=Coeff[PolynomialRemainder[x^m*(d+e*x^2)^q,a+b*x^2+c*x^4,x],x,2]},
  x*(a+b*x^2+c*x^4)^(p+1)*(a*b*g-f*(b^2-2*a*c)-c*(b*f-2*a*g)*x^2)/(2*a*(p+1)*(b^2-4*a*c)) +
   1/(2*a*(p+1)*(b^2-4*a*c))*Int[x^m*(a+b*x^2+c*x^4)^(p+1)*
     Simp[ExpandToSum[2*a*(p+1)*(b^2-4*a*c)*x^(-m)*PolynomialQuotient[x^m*(d+e*x^2)^q,a+b*x^2+c*x^4,x]+
       (b^2*f*(2*p+3)-2*a*c*f*(4*p+5)-a*b*g)*x^(-m)+c*(4*p+7)*(b*f-2*a*g)*x^(2-m),x],x],x]] /;
FreeQ[{a,b,c,d,e},x] && NeQ[b^2-4*a*c,0] && LtQ[p,-1] && IGtQ[q,1] && ILtQ[m/2,0]
```

14: $\int (f x)^m (d+e x^2)^q (a+b x^2+c x^4)^p \, dx$ when $b^2-4ac \neq 0 \wedge (p \in \mathbb{Z}^+ \vee q \in \mathbb{Z}^+ \vee (m \mid q) \in \mathbb{Z})$

Derivation: Algebraic expansion

Rule 1.2.2.4.14: If $b^2-4ac \neq 0 \wedge (p \in \mathbb{Z}^+ \vee q \in \mathbb{Z}^+ \vee (m \mid q) \in \mathbb{Z})$, then

$$\int (f x)^m (d+e x^2)^q (a+b x^2+c x^4)^p \, dx \longrightarrow \int \text{ExpandIntegrand}\left[(f x)^m (d+e x^2)^q (a+b x^2+c x^4)^p, x\right] dx$$

Program code:

```
Int[(f_.*x_)^m_.*(d_+e_.*x_^2)^q_.*(a_+b_.*x_^2+c_.*x_^4)^p_.,x_Symbol] :=
  Int[ExpandIntegrand[(f*x)^m*(d+e*x^2)^q*(a+b*x^2+c*x^4)^p,x],x] /;
FreeQ[{a,b,c,d,e,f,m,p,q},x] && NeQ[b^2-4*a*c,0] && (IGtQ[p,0] || IGtQ[q,0] || IntegersQ[m,q])
```

```
Int[(f_.*x_)^m_.*(d_+e_.*x_^2)^q_.*(a_+c_.*x_^4)^p_.,x_Symbol] :=
  Int[ExpandIntegrand[(f*x)^m*(d+e*x^2)^q*(a+c*x^4)^p,x],x] /;
FreeQ[{a,c,d,e,f,m,p,q},x] && (IGtQ[p,0] || IGtQ[q,0] || IntegersQ[m,q])
```

15: $\int (f\,x)^m \left(d+e\,x^2\right)^q \left(a+c\,x^4\right)^p dx$ when $p\notin\mathbb{Z}\,\land\,q\in\mathbb{Z}^-$

Derivation: Algebraic expansion

Basis: If $q\in\mathbb{Z}$, then $\left(d+e\,x^2\right)^q == \left(\frac{d}{d^2-e^2\,x^4}-\frac{e\,x^2}{d^2-e^2\,x^4}\right)^{-q}$

Note: Resulting integrands are of the form $x^m \left(a+b\,x^2\right)^p \left(c+d\,x^2\right)^q$ which are integrable.

Rule 1.2.2.4.15: If $p\notin\mathbb{Z}\,\land\,q\in\mathbb{Z}^-$, then

$$\int (f\,x)^m \left(d+e\,x^2\right)^q \left(a+c\,x^4\right)^p dx \rightarrow \frac{(f\,x)^m}{x^m}\int x^m \left(a+c\,x^4\right)^p \text{ExpandIntegrand}\left[\left(\frac{d}{d^2-e^2\,x^4}-\frac{e\,x^2}{d^2-e^2\,x^4}\right)^{-q},\,x\right] dx$$

Program code:

```
Int[(f_.*x_)^m_.*(d_+e_.*x_^2)^q_*(a_+c_.*x_^4)^p_,x_Symbol] :=
  (f*x)^m/x^m*Int[ExpandIntegrand[x^m*(a+c*x^4)^p,(d/(d^2-e^2*x^4)-e*x^2/(d^2-e^2*x^4))^(-q),x],x] /;
FreeQ[{a,c,d,e,f,m,p},x] && Not[IntegerQ[p]] && ILtQ[q,0]
```

U: $\int (f\,x)^m \left(d+e\,x^2\right)^q \left(a+b\,x^2+c\,x^4\right)^p dx$

Rule 1.2.2.4.U:

$$\int (f\,x)^m \left(d+e\,x^2\right)^q \left(a+b\,x^2+c\,x^4\right)^p dx \rightarrow \int (f\,x)^m \left(d+e\,x^2\right)^q \left(a+b\,x^2+c\,x^4\right)^p dx$$

Program code:

```
Int[(f_.*x_)^m_.*(d_+e_.*x_^2)^q_.*(a_+b_.*x_^2+c_.*x_^4)^p_.,x_Symbol] :=
  Unintegrable[(f*x)^m*(d+e*x^2)^q*(a+b*x^2+c*x^4)^p,x] /;
FreeQ[{a,b,c,d,e,f,m,p,q},x]
```

```
Int[(f_.*x_)^m_.*(d_+e_.*x_^2)^q_.*(a_+c_.*x_^4)^p_.,x_Symbol] :=
  Unintegrable[(f*x)^m*(d+e*x^2)^q*(a+c*x^4)^p,x] /;
FreeQ[{a,c,d,e,f,m,p,q},x]
```